In [1]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
import MeshCat as mc
```

```
  Activating project at `c:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW3_S25`
```

In [2]:
```
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","cartpole_animation.jl"))
```

```
animate_cartpole (generic function with 1 method)
```

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q1: Direct Collocation (DIRCOL) for a Cart Pole (30 pts)

We are now going to start working with the NonLinear Program (NLP) Solver IPOPT to solve some trajectory optimization problems. First we will demonstrate how this works for simple optimization problems (not trajectory optimization). The interface that we have setup for IPOPT is the following:

$$
\begin{aligned}
\min_{x} \quad & \ell(x) && \text{cost function} \\
\text{st} \quad & c_{eq}(x) = 0 && \text{equality constraint} \\
& c_L \leq c_{ineq}(x) \leq c_U && \text{inequality constraint} \\
& x_L \leq x \leq x_U && \text{primal bound constraint}
\end{aligned}
$$

where $\ell(x)$ is our objective function, $c_{eq}(x) = 0$ is our equality constraint, $c_L \leq c_{ineq}(x) \leq c_U$ is our bound inequality constraint, and $x_L \leq x \leq x_U$ is a bound constraint on our primal variable x.

Part A: Solve an LP with IPOPT (5 pts)

To demonstrate this, we are going to ask you to solve a simple Linear Program (LP):

$$
\begin{aligned}
\min_{x} \quad & q^T x \\
\text{st} \quad & Ax = b \\
& Gx \leq h
\end{aligned}
$$

Your job will be to transform this problem into the form shown above and solve it with IPOPT. To help you interface with IPOPT, we have created a function `fmincon` for you. Below is the docstring for this function that details all of the inputs.

In [3]:
```
"""
x = fmincon(cost,equality_constraint,inequality_constraint,x_l,x_u,c_l,c_u,x0,params,diff_type)

This function uses IPOPT to minimize an objective function

`cost(params, x)`

With the following three constraints:

`equality_constraint(params, x) = 0`
`c_l <= inequality_constraint(params, x) <= c_u`
`x_l <= x <= x_u`

Note that the constraint functions should return vectors.

Problem specific parameters should be loaded into params::NamedTuple (things like
cost weights, dynamics parameters, etc.).

args:
    cost::Function                    - objective function to be minimzed (returns scalar)
    equality_constraint::Function     - c_eq(params, x) == 0
    inequality_constraint::Function   - c_l <= c_ineq(params, x) <= c_u
```

```
    x_l::Vector                       - x_l <= x <= x_u
    x_u::Vector                       - x_l <= x <= x_u
    c_l::Vector                       - c_l <= c_ineq(params, x) <= x_u
    c_u::Vector                       - c_l <= c_ineq(params, x) <= x_u
    x0::Vector                        - initial guess
    params::NamedTuple                - problem parameters for use in costs/constraints
    diff_type::Symbol                 - :auto for ForwardDiff, :finite for FiniteDiff
    verbose::Bool                     - true for IPOPT output, false for nothing

optional args:
    tol                               - optimality tolerance
    c_tol                             - constraint violation tolerance
    max_iters                         - max iterations
    verbose                           - verbosity of IPOPT

outputs:
    x::Vector                         - solution

You should try and use :auto for your `diff_type` first, and only use :finite if you
absolutely cannot get ForwardDiff to work.

This function will run a few basic checks before sending the problem off to IPOPT to
solve. The outputs of these checks will be reported as the following:

---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------

If you're getting stuck during the testing of one of the derivatives, try switching
to FiniteDiff.jl by setting diff_type = :finite.
"""
```

```
"x = fmincon(cost,equality_constraint,inequality_constraint,x_l,x_u,c_l,c_u,x0,params,diff_type)\n\nThis functio
n uses IPOPT to minimize an objective function \n\n`cost(params, x)` \n\nWith the following three constraints: \
n\n`equality_constraint(params, x) = 0`\n`c_l <= inequal" ⋯ 1899 bytes ⋯ "nt Jacobian----------------\n---------
successfully compiled both derivatives-----\n---------IPOPT beginning solve----------------------\n\nIf you're g
etting stuck during the testing of one of the derivatives, try switching \nto FiniteDiff.jl by setting diff_type
 = :finite. \n"
```

In [4]:

```
@testset "solve LP with IPOPT" begin

    LP = jldopen(joinpath(@__DIR__,"utils","random_LP.jld2"))

    params = (q = LP["q"], A = LP["A"], b = LP["b"], G = LP["G"], h = LP["h"])

    # return a scalar
    function cost(params, x)::Real
        # TODO: create cost function with params and x
        return dot(x, params.q)
    end

    # return a vector
    function equality_constraint(params, x)::Vector
        # TODO: create equality constraint function with params and x
        return params.A * x - params.b
    end

    # return a vector
    function inequality_constraint(params, x)::Vector
        # TODO: create inequality constraint function with params and x
        return params.G * x - params.h
    end

    # TODO: primal bounds
    # you may use Inf, like Inf*ones(10) for a vector of positive infinity
    x_l = -Inf * ones(length(params.q))
    x_u = Inf * ones(length(params.q))

    # TODO: inequality constraint bounds
    c_l = -Inf * ones(size(params.G, 1))
    c_u = zeros(size(params.G, 1))

    # initial guess
    x0 = randn(length(params.q))

    diff_type = :auto   # use ForwardDiff.jl
#     diff_type = :finite # use FiniteDiff.jl

    x = fmincon(cost, equality_constraint, inequality_constraint,
                x_l, x_u, c_l, c_u, x0, params, diff_type;
```

```julia
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true);


    @test isapprox(x, [-0.44289, 0, 0, 0.19214, 0, 0, -0.109095,
                       -0.43221, 0, 0, 0.44289, 0, 0, 0.192142,
                       0, 0, 0.10909, 0.432219, 0, 0], atol = 1e-3)

end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------

******************************************************************************
This program contains Ipopt, a library for large-scale nonlinear optimization.
 Ipopt is released as open source code under the Eclipse Public License (EPL).
         For more information visit https://github.com/coin-or/Ipopt
******************************************************************************

This is Ipopt version 3.14.17, running with linear solver MUMPS 5.7.3.

Number of nonzeros in equality constraint Jacobian...:       80
Number of nonzeros in inequality constraint Jacobian.:      400
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:       20
                     variables with only lower bounds:        0
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:        4
Total number of inequality constraints...............:       20
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:       20

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0 -2.3857109e+00 2.45e+00 3.33e-01   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  1.5619453e+00 4.71e-01 1.21e+00  -1.0 2.14e+00    -  4.09e-01 8.08e-01h  1
   2  3.4593804e+00 1.11e-16 9.88e-07  -0.8 6.94e-01    -  1.00e+00 1.00e+00f  1
   3  1.5015602e+00 6.94e-17 7.53e-09  -3.0 4.84e-01    -  1.00e+00 8.13e-01f  1
   4  1.2732586e+00 2.22e-16 2.14e-08  -2.5 1.02e-01    -  9.75e-01 8.33e-01f  1
   5  1.1832352e+00 2.22e-16 1.21e-02  -3.7 3.90e-02    -  9.28e-01 9.96e-01f  1
   6  1.1767352e+00 1.11e-16 2.07e-03  -5.3 6.22e-03    -  1.00e+00 9.12e-01f  1
   7  1.1763558e+00 1.11e-16 1.25e-12  -6.8 1.25e-04    -  1.00e+00 9.87e-01f  1

Number of Iterations....: 7

                                   (scaled)                 (unscaled)
Objective...............:   1.1763558478713843e+00    1.1763558478713843e+00
Dual infeasibility......:   1.2495005030643824e-12    1.2495005030643824e-12
Constraint violation....:   1.1102230246251565e-16    1.1102230246251565e-16
Variable bound violation:   0.0000000000000000e+00    0.0000000000000000e+00
Complementarity.........:   5.5224373681030012e-07    5.5224373681030012e-07
Overall NLP error.......:   5.5224373681030012e-07    5.5224373681030012e-07


Number of objective function evaluations             = 8
Number of objective gradient evaluations             = 8
Number of equality constraint evaluations            = 8
Number of inequality constraint evaluations          = 8
Number of equality constraint Jacobian evaluations   = 8
Number of inequality constraint Jacobian evaluations = 8
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 1.668

EXIT: Optimal Solution Found.
Test Summary:       | Pass  Total  Time
solve LP with IPOPT |    1      1  6.7s
Test.DefaultTestSet("solve LP with IPOPT", Any[], 1, false, false, true, 1.7428244741851e9, 1.742824480841e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW3_S25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_W5sZmlsZQ==.jl")
```

Part B: Cart Pole Swingup (20 pts)

We are now going to solve for a cartpole swingup. The state for the cartpole is the following:

$$x = [p, \theta, \dot{p}, \dot{\theta}]^T$$

Where p and θ can be seen in the graphic `cartpole.png`.

No description has been provided for this image

where we start with the pole in the down position ($\theta = 0$), and we want to use the horizontal force on the cart to drive the pole to the up position ($\theta = \pi$).

$$
\begin{aligned}
\min_{x_{1:N}, u_{1:N-1}} \quad & \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{goal})^T Q (x_i - x_{goal}) + \frac{1}{2} u_i^T R u_i \right] + \frac{1}{2}(x_N - x_{goal})^T Q_f (x_N - x_{goal}) \\
\text{st} \quad & x_1 = x_{\text{IC}} \\
& x_N = x_{goal} \\
& f_{hs}(x_i, x_{i+1}, u_i, dt) = 0 \quad \text{for } i = 1, 2, \ldots, N-1 \\
& -10 \leq u_i \leq 10 \quad \text{for } i = 1, 2, \ldots, N-1
\end{aligned}
$$

Where $x_{IC} = [0, 0, 0, 0]$, and $x_{goal} = [0, \pi, 0, 0]$, and $f_{hs}(x_i, x_{i+1}, u_i)$ is the implicit integrator residual for Hermite Simpson (see HW1Q1 to refresh on this). Note that while Zac used a first order hold (FOH) on the controls in class (meaning we linearly interpolate controls between time steps), we are using a zero-order hold (ZOH) in this assignment. This means that each control u_i is held constant for the entirety of the timestep.

In [5]:
```julia
# cartpole
function dynamics(params::NamedTuple, x::Vector, u)
    # cartpole ODE, parametrized by params.

    # cartpole physical parameters
    mc, mp, l = params.mc, params.mp, params.l
    g = 9.81

    q = x[1:2]
    qd = x[3:4]

    s = sin(q[2])
    c = cos(q[2])

    H = [mc+mp mp*l*c; mp*l*c mp*l^2]
    C = [0 -mp*qd[2]*l*s; 0 0]
    G = [0, mp*g*l*s]
    B = [1, 0]

    qdd = -H\(C*qd + G - B*u[1])
    xdot = [qd;qdd]
    return xdot

end
function hermite_simpson(params::NamedTuple, x1::Vector, x2::Vector, u, dt::Real)::Vector
    # TODO: input hermite simpson implicit integrator residual
    x_mid = 0.5 * (x1 + x2)
    u_mid = u

    f1 = dynamics(params, x1, u)
    f2 = dynamics(params, x2, u)
    f_mid = dynamics(params, x_mid, u_mid)

    residual = x2 - x1 - (dt / 6) * (f1 + 4 * f_mid + f2)
    return residual
end
```

```
hermite_simpson (generic function with 1 method)
```

To solve this problem with IPOPT and `fmincon`, we are going to concatenate all of our x's and u's into one vector:

$$Z = \begin{bmatrix} x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N \cdot nx + (N-1) \cdot nu}$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z.

It is also worth noting that while there are inequality constraints present ($-10 \leq u_i \leq 10$), we do not need a specific `inequality_constraints` function as an input to `fmincon` since these are just bounds on the primal (Z) variable. You should use primal bounds in `fmincon` to capture these constraints.

In [6]:
```
function create_idx(nx,nu,N)
    # This function creates some useful indexing tools for Z
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]

    # Feel free to use/not use anything here.


    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function cartpole_cost(params::NamedTuple, Z::Vector)::Real
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf

    # TODO: input cartpole LQR cost
    xg = params.xg
    J = 0
    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]

        J += 0.5*(xi-xg)'*Q*(xi-xg) + 0.5*ui'*R*ui


    end

    # dont forget terminal cost
    J += 0.5*(Z[idx.x[N]]-xg)'*Qf*(Z[idx.x[N]]-xg)

    return J
end

function cartpole_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt

    # TODO: create dynamics constraints using hermite simpson

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)

    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]
        xip1 = Z[idx.x[i+1]]

        # TODO: hermite simpson
        c[idx.c[i]] = hermite_simpson(params, xi, xip1, ui, dt)
    end
    return c
```

```
end

function cartpole_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # TODO: return all of the equality constraints
    num_dyn_constr = idx.nc
    num_state = length(xic)

    eq_constraint = similar(Z, num_dyn_constr + 2 * num_state)
    eq_constraint[1:num_dyn_constr]  = cartpole_dynamics_constraints(params, Z)
    eq_constraint[num_dyn_constr .+ (1:num_state)] = Z[idx.x[1]] - xic
    eq_constraint[num_dyn_constr .+ num_state .+ (1:num_state)] = Z[idx.x[N]] - xg

    return eq_constraint
end

function solve_cartpole_swingup(;verbose=true)

    # problem size
    nx = 4
    nu = 1
    dt = 0.05
    tf = 2.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # LQR cost
    Q = diagm(ones(nx))
    R = 0.1*diagm(ones(nu))
    Qf = 10*diagm(ones(nx))

    # indexing
    idx = create_idx(nx,nu,N)

    # initial and goal states
    xic = [0, 0, 0, 0]
    xg = [0, pi, 0, 0]

    # load all useful things into params
    params = (Q = Q, R = R, Qf = Qf, xic = xic, xg = xg, dt = dt, N = N, idx = idx,mc = 1.0, mp = 0.2, l = 0.5)

    # TODO: primal bounds
    x_l = fill(-Inf, idx.nz)
    x_u = fill(Inf, idx.nz)

    for i in 1:(N-1)
        x_l[idx.u[i]] .= -10
        x_u[idx.u[i]] .= 10
    end



    # inequality constraint bounds (this is what we do when we have no inequality constraints)
    c_l = zeros(0)
    c_u = zeros(0)
    function inequality_constraint(params, Z)
        return zeros(eltype(Z), 0)
    end

    # initial guess
    z0 = 0.001*randn(idx.nz)

    # choose diff type (try :auto, then use :finite if :auto doesn't work)
    diff_type = :auto
#     diff_type = :finite


    Z = fmincon(cartpole_cost,cartpole_equality_constraint,inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = verbose)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    return X, U, t_vec, params
end

@testset "cartpole swingup" begin

    X, U, t_vec = solve_cartpole_swingup(verbose=true)
```

```
    # --------------testing------------------
    @test isapprox(X[1],zeros(4), atol = 1e-4)
    @test isapprox(X[end], [0,pi,0,0], atol = 1e-4)
    Xm = hcat(X...)
    Um = hcat(U...)

    # --------------plotting-----------------
    display(plot(t_vec, Xm', label = ["p" "θ" "ṗ" "θ̇"], xlabel = "time (s)", title = "State Trajectory"))
    display(plot(t_vec[1:end-1],Um',label="",xlabel = "time (s)", ylabel = "u",title = "Controls"))

    # meshcat animation
    display(animate_cartpole(X, 0.05))

end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------
This is Ipopt version 3.14.17, running with linear solver MUMPS 5.7.3.

Number of nonzeros in equality constraint Jacobian...:    34272
Number of nonzeros in inequality constraint Jacobian.:        0
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      204
                     variables with only lower bounds:        0
                variables with lower and upper bounds:       40
                     variables with only upper bounds:        0
Total number of equality constraints.................:      168
Total number of inequality constraints...............:        0
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.4673376e+02 3.14e+00 5.88e-04   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  2.7532763e+02 2.38e+00 7.87e+00  -5.0 1.28e+01    -  4.90e-01 2.43e-01h  3
   2  2.9864670e+02 2.16e+00 1.01e+01  -0.6 1.05e+01    -  6.07e-01 9.29e-02h  4
   3  3.3529994e+02 1.87e+00 1.39e+01  -0.4 1.29e+01    -  6.47e-01 1.34e-01h  3
   4  3.7253594e+02 1.61e+00 2.06e+01  -0.5 1.19e+01    -  8.67e-01 1.40e-01h  3
   5  4.2132582e+02 1.33e+00 2.71e+01  -0.8 9.95e+00    -  1.00e+00 1.74e-01h  3
   6  4.4528033e+02 1.20e+00 3.17e+01   0.2 1.84e+01    -  6.26e-01 9.53e-02h  3
   7  4.7644890e+02 1.07e+00 3.52e+01   0.2 1.78e+01    -  6.12e-01 1.10e-01h  3
   8  5.1226243e+02 9.45e-01 3.88e+01   0.3 2.24e+01    -  6.25e-01 1.16e-01h  3
   9  5.2184419e+02 8.53e-01 3.82e+01   0.3 1.16e+01    -  8.77e-01 9.66e-02h  3
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  5.1714390e+02 6.91e-01 4.59e+01   0.4 2.63e+01    -  5.10e-01 1.90e-01f  2
  11  5.1195320e+02 6.30e-01 4.81e+01   0.4 2.18e+01    -  3.40e-01 8.75e-02f  3
  12  5.0987020e+02 5.65e-01 5.09e+01   0.4 2.91e+01    -  8.74e-01 1.04e-01h  3
  13  5.4466863e+02 3.39e-01 7.18e+01   0.4 2.01e+01    -  3.43e-01 3.99e-01h  1
  14  5.4654300e+02 2.19e-01 7.83e+01   0.4 1.60e+01    -  4.58e-01 3.55e-01h  1
  15  5.4679044e+02 1.81e-01 7.79e+01   0.4 1.14e+01    -  7.74e-01 1.73e-01h  1
  16  5.4918562e+02 1.26e-01 8.23e+01   0.6 1.22e+01    -  8.35e-01 3.76e-01h  1
  17  5.4742385e+02 1.01e-01 8.02e+01   0.6 1.01e+01    -  6.49e-01 5.41e-01h  1
  18  5.3273194e+02 9.88e-02 5.33e+01   0.3 6.68e+00    -  9.36e-01 9.82e-01f  1
  19  5.0613753e+02 3.68e-02 2.01e+01   0.1 1.93e+00    -  9.90e-01 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  4.8660423e+02 5.24e-02 1.93e+01  -0.8 1.15e+01    -  3.38e-01 2.42e-01f  1
  21  4.7098149e+02 1.50e-01 3.89e+01   0.1 3.57e+01    -  7.11e-01 1.52e-01f  1
  22  4.6153523e+02 1.19e-01 6.27e+01   0.3 1.15e+01    -  8.72e-01 5.52e-01f  1
  23  4.5176371e+02 5.94e-03 2.86e+01   0.2 2.32e+00    -  9.53e-01 1.00e+00f  1
  24  4.4535299e+02 5.07e-02 2.43e+01  -0.2 5.36e+00    -  9.78e-01 1.00e+00f  1
  25  4.4265386e+02 7.32e-02 3.04e+01  -0.9 1.95e+01    -  3.01e-01 2.99e-01f  2
  26  4.4186075e+02 6.08e-02 3.41e+01  -0.0 8.63e+00    -  7.48e-01 6.83e-01f  1
  27  4.3326599e+02 1.18e-02 3.18e+01  -0.5 2.65e+00    -  9.91e-01 1.00e+00f  1
  28  4.3126032e+02 1.19e-03 2.16e+01  -1.0 9.39e-01    -  9.98e-01 1.00e+00f  1
  29  4.3044249e+02 2.18e-02 2.67e+01  -1.4 5.18e+00    -  9.99e-01 6.22e-01f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  4.2926101e+02 1.31e-02 2.79e+01  -0.9 7.41e+00    -  1.00e+00 1.00e+00f  1
  31  4.2671974e+02 1.50e-02 2.22e+01  -1.0 2.96e+00    -  1.00e+00 1.00e+00f  1
  32  4.2519312e+02 3.68e-02 2.05e+01  -1.4 3.93e+00    -  9.99e-01 1.00e+00f  1
  33  4.2856258e+02 4.41e-02 3.60e+01  -0.5 1.91e+01    -  8.95e-01 7.52e-01F  1
  34  4.2167135e+02 9.50e-03 2.99e+01  -0.6 6.88e+00    -  1.00e+00 1.00e+00f  1
  35  4.1898529e+02 5.80e-03 1.60e+01  -1.0 2.09e+00    -  9.92e-01 8.32e-01f  1
  36  4.1753240e+02 1.08e-02 2.14e+01  -1.4 5.60e+00    -  1.00e+00 6.52e-01f  1
  37  4.1694284e+02 2.33e-02 2.25e+01  -1.7 4.59e+00    -  1.00e+00 8.08e-01f  1
  38  4.1475942e+02 7.14e-03 1.72e+01  -2.1 2.44e+00    -  1.00e+00 8.30e-01f  1
  39  4.1596675e+02 2.48e-02 1.14e+01  -1.0 2.98e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
```

```
  40  4.1475131e+02 2.61e-02 1.02e+01  -1.0 1.53e+00    -  1.00e+00 1.00e+00f  1
  41  4.1323805e+02 9.33e-03 8.88e+00  -1.0 1.93e+00    -  9.65e-01 1.00e+00f  1
  42  4.1282719e+02 1.50e-02 1.29e+01  -1.2 1.93e+01    -  9.82e-01 8.60e-02f  3
  43  4.1338860e+02 2.05e-02 1.80e+01  -0.6 1.44e+01    -  1.00e+00 8.84e-02f  3
  44  4.0808006e+02 1.69e-02 1.41e+01  -0.7 8.31e+00    -  1.00e+00 1.00e+00F  1
  45  4.0826854e+02 1.11e-04 1.53e+00  -1.2 5.86e-01    -  1.00e+00 1.00e+00h  1
  46  4.0687195e+02 4.50e-03 5.53e+00  -1.5 5.52e+00    -  9.60e-01 4.73e-01f  1
  47  4.0698978e+02 3.30e-02 2.73e+01  -0.9 1.20e+01    -  9.18e-01 1.00e+00F  1
  48  4.0458935e+02 8.38e-03 1.83e+01  -1.0 1.65e+00    -  1.00e+00 7.92e-01f  1
  49  4.0308005e+02 1.74e-04 1.30e+01  -1.4 1.02e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  50  4.0246115e+02 8.89e-03 1.67e+01  -1.9 7.49e+00    -  1.00e+00 3.11e-01f  1
  51  3.9920523e+02 2.94e-03 1.76e+01  -1.5 5.43e+00    -  1.00e+00 1.00e+00F  1
  52  4.1088434e+02 1.95e-03 1.04e+01  -0.5 6.61e+00    -  8.43e-01 1.00e+00H  1
  53  3.9894262e+02 9.92e-03 2.28e+01  -0.6 4.65e+00    -  1.00e+00 1.00e+00f  1
  54  3.9835535e+02 6.07e-04 8.26e+00  -0.7 1.75e+00    -  9.81e-01 1.00e+00f  1
  55  3.9564920e+02 3.35e-03 1.41e+01  -1.5 4.34e+00    -  9.86e-01 1.00e+00F  1
  56  3.9553486e+02 7.15e-04 1.16e+01  -1.5 5.61e+00    -  9.83e-01 7.98e-01H  1
  57  3.9537308e+02 3.65e-04 3.58e+00  -1.5 6.01e-01    -  1.00e+00 1.00e+00f  1
  58  3.9518589e+02 9.47e-04 3.98e+00  -2.3 7.39e-01    -  1.00e+00 1.00e+00f  1
  59  3.9636344e+02 1.66e-02 1.53e+01  -0.6 8.70e+01    -  6.85e-01 3.32e-02f  2
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  60  3.9631539e+02 2.59e-03 9.71e+00  -0.9 2.04e+00    -  8.33e-01 1.00e+00f  1
  61  3.9537661e+02 5.69e-04 8.75e+00  -0.9 2.03e+00    -  1.00e+00 1.00e+00f  1
  62  3.9533589e+02 2.55e-04 7.05e+00  -0.9 4.21e-01    -  1.00e+00 1.00e+00h  1
  63  3.9443279e+02 1.37e-03 3.28e+00  -1.3 2.17e+00    -  1.00e+00 1.00e+00F  1
  64  3.9435413e+02 1.59e-04 1.32e+00  -1.9 4.71e-01    -  1.00e+00 1.00e+00h  1
  65  3.9423722e+02 4.56e-05 2.40e-01  -3.1 2.66e-01    -  1.00e+00 1.00e+00f  1
  66  3.9422920e+02 2.85e-06 5.92e-02  -4.6 5.16e-02    -  1.00e+00 9.88e-01h  1
  67  3.9422857e+02 6.00e-08 1.03e-03  -5.9 1.29e-02    -  1.00e+00 9.86e-01h  1
  68  3.9422855e+02 1.24e-08 4.26e-04  -8.0 3.21e-03    -  1.00e+00 9.98e-01h  1
  69  3.9422855e+02 4.13e-10 1.02e-04 -10.0 2.08e-03    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  70  3.9422855e+02 6.10e-11 5.69e-06 -11.0 3.79e-04    -  1.00e+00 1.00e+00h  1
  71  3.9422855e+02 4.31e-12 5.45e-06 -11.0 5.50e-05    -  1.00e+00 1.00e+00h  1
  72  3.9422855e+02 1.43e-13 9.90e-07 -11.0 1.59e-05    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 72

                                   (scaled)                 (unscaled)
Objective...............:   3.9422855324842999e+02    3.9422855324842999e+02
Dual infeasibility......:   9.8978321193548184e-07    9.8978321193548184e-07
Constraint violation....:   1.4299672557172016e-13    1.4299672557172016e-13
Variable bound violation:   9.9997189195732972e-08    9.9997189195732972e-08
Complementarity.........:   1.0002479588953149e-11    1.0002479588953149e-11
Overall NLP error.......:   9.8978321193548184e-07    9.8978321193548184e-07


Number of objective function evaluations             = 154
Number of objective gradient evaluations             = 73
Number of equality constraint evaluations            = 154
Number of inequality constraint evaluations          = 0
Number of equality constraint Jacobian evaluations   = 73
Number of inequality constraint Jacobian evaluations = 0
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 7.003

EXIT: Optimal Solution Found.
```






```
┌ Info: Listening on: 127.0.0.1:8701, thread id: 1
└ @ HTTP.Servers C:\Users\barat\.julia\packages\HTTP\4AUPl\src\Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8701
└ @ MeshCat C:\Users\barat\.julia\packages\MeshCat\9QrxD\src\visualizer.jl:43
```

```
Test Summary:     | Pass  Total   Time
cartpole swingup |    2      2  22.5s
Test.DefaultTestSet("cartpole swingup", Any[], 2, false, false, true, 1.742824481456e9, 1.742824503921e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW3_S25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X12sZmlsZQ==.jl")
```

Part C: Track DIRCOL Solution (5 pts)

Now, similar to HW2 Q2 Part C, we are taking a solution X and U from DIRCOL, and we are going to track the trajectory with TVLQR to account for model mismatch. While we used hermite-simpson integration for the dynamics constraints in DIRCOL, we are going to use RK4 for this simulation. Remember to clamp your control to be within the control bounds.

In [7]:
```julia
function rk4(params::NamedTuple, x::Vector,u,dt::Float64)
    # vanilla RK4
    k1 = dt*dynamics(params, x, u)
    k2 = dt*dynamics(params, x + k1/2, u)
    k3 = dt*dynamics(params, x + k2/2, u)
    k4 = dt*dynamics(params, x + k3, u)
    x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end
```

```julia
@testset "track cartpole swingup with TVLQR" begin

    X_dircol, U_dircol, t_vec, params_dircol = solve_cartpole_swingup(verbose = false)

    N = length(X_dircol)
    dt = params_dircol.dt
    x0 = X_dircol[1]

    # TODO: use TVLQR to generate K's

    nx = length(X_dircol[1]) # number of states
    nu = length(U_dircol[1]) # number of controls

    # use this for TVLQR tracking cost
    Q = diagm([1,1,.05,.1])
    Qf = 100*Q
    R = 0.01*diagm(ones(1))

    P = [zeros(nx,nx) for i = 1:N]
    K = [zeros(nu,nx) for i = 1:N-1]
    P[N] = deepcopy(Qf)
    ugoal = [0]

    for i in N-1:-1:1
        A = ForwardDiff.jacobian(x -> rk4(params_dircol, x, U_dircol[i], dt), X_dircol[i])
        B = ForwardDiff.jacobian(u -> rk4(params_dircol, X_dircol[i], u, dt), U_dircol[i])

        K[i] = (R + B'*P[i+1]*B)\(B'*P[i+1]*A)
        P[i] = Q + A'*P[i+1]*(A - B*K[i])
    end

    # simulation
    Xsim = [zeros(nx) for i = 1:N]
    Usim = [zeros(nu) for i = 1:(N-1)]
    Xsim[1] = 1*x0

    # here are the real parameters (different than the one we used for DIRCOL)
    # this model mismatch is what's going to require the TVLQR controller to track
    # the trajectory successfully.
    params_real = (mc = 1.05, mp = 0.21, l = 0.48)

    # TODO: simulate closed loop system with both feedforward and feedback control
    # feedforward - the U_dircol controls that we solved for using dircol
    # feedback - the TVLQR controls
    for i = 1:(N-1)
        # add controller and simulation step
        Usim[i] = U_dircol[i] - K[i]*(Xsim[i] - X_dircol[i])
        Usim[i] .= clamp.(Usim[i], -10, 10)
        Xsim[i+1] = rk4(params_real, Xsim[i], Usim[i], dt)
    end


    # -----------------testing----------------------
    xn = Xsim[N]
    @test norm(xn)>0
    @test 1e-6<norm(xn - X_dircol[end])<.8
    @test abs(abs(rad2deg(xn[2])) - 180) < 5 # within 5 degrees
    @test maximum(norm.(Usim,Inf)) <= (10 + 1e-3)

    # -----------------plotting---------------------
    Xm = hcat(Xsim...)
    Xbarm = hcat(X_dircol...)
    plot(t_vec,Xbarm',ls=:dash, label = "",lc = [:red :green :blue :black])
    display(plot!(t_vec,Xm',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "x",
                 label = ["p" "θ" "ṗ" "θ̇"],lc = [:red :green :blue :black]))

    Um = hcat(Usim...)
    Ubarm = hcat(U_dircol...)
    plot(t_vec[1:end-1],Ubarm',ls=:dash,lc = :blue, label = "")
    display(plot!(t_vec[1:end-1],Um',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "u",lc = :blue, label = ""))

    # ----------------animate-----------------------
    display(animate_cartpole(Xsim, 0.05))

end
```






```
┌ Info: Listening on: 127.0.0.1:8704, thread id: 1
└ @ HTTP.Servers C:\Users\barat\.julia\packages\HTTP\4AUPl\src\Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8704
└ @ MeshCat C:\Users\barat\.julia\packages\MeshCat\9QrxD\src\visualizer.jl:43
```

```
Test Summary:                      | Pass  Total   Time
track cartpole swingup with TVLQR |    4      4  11.0s
Test.DefaultTestSet("track cartpole swingup with TVLQR", Any[], 4, false, false, true, 1.742824503969e9, 1.742824514975e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW3_S25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X14sZmlsZQ==.jl")
```

In [1]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()

import MathOptInterface as MOI
import Ipopt
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
import MeshCat as mc
using Printf
```

```
  Activating project at `c:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW3
_S25`
```

Q2: iLQR (30 pts)

In this problem, we are going to use iLQR to solve a trajectory optimization for a 6DOF quadrotor. This problem we will use a cost function to motivate the quadrotor to follow a specified aerobatic manuever. The continuous time dynamics of the quadrotor are detailed in `quadrotor.jl`, with the state being the following:

$x = [r, v, {}^N p^B, \omega]$ where $r \in \mathbb{R}^3$ is the position of the quadrotor in the world frame (N), $v \in \mathbb{R}^3$ is the velocity of the quadrotor in the world frame (N), ${}^N p^B \in \mathbb{R}^3$ is the Modified Rodrigues Parameter (MRP) that is used to denote the attitude of the quadrotor, and $\omega \in \mathbb{R}^3$ is the angular velocity of the quadrotor expressed in the body frame (B). By denoting the attitude of the quadrotor with a MRP instead of a quaternion or rotation matrix, we have to be careful to avoid any scenarios where the MRP will approach it's singularity at 360 degrees of rotation. For the manuever planned in this problem, the MRP will be sufficient.

The dynamics of the quadrotor are discretized with `rk4`, resulting in the following discrete time dynamics function:

In [2]:
```
include(joinpath(@__DIR__, "utils","quadrotor.jl"))

function discrete_dynamics(params::NamedTuple, x::Vector, u, k)
    # discrete dynamics
    # x - state
    # u - control
    # k - index of trajectory
    # dt comes from params.model.dt
    return rk4(params.model, quadrotor_dynamics, x, u, params.model.dt)
end
```

```
discrete_dynamics (generic function with 1 method)
```

Part A: iLQR for a quadrotor (25 pts)

iLQR is used to solve optimal control problems of the following form:

$$\begin{aligned}
\min_{x_{1:N}, u_{1:N-1}} \quad & \left[\sum_{i=1}^{N-1} \ell(x_i, u_i)\right] + \ell_N(x_N) \\
\text{st} \quad & x_1 = x_{IC} \\
& x_{k+1} = f(x_k, u_k) \quad \text{for } i = 1, 2, \ldots, N-1
\end{aligned}$$

where x_{IC} is the inital condition, $x_{k+1} = f(x_k, u_k)$ is the discrete dynamics function, $\ell(x_i, u_i)$ is the stage cost, and $\ell_N(x_N)$ is the terminal cost. Since this optimization problem can be non-convex, there is no guarantee of convergence to a global optimum, or even convergene rates to a local optimum, but in practice we will see that it can work very well.

For this problem, we are going to use a simple cost function consisting of the following stage cost:

$$\ell(x_i, u_i) = \frac{1}{2}(x_i - x_{ref,i})^T Q (x_i - x_{ref,i}) + \frac{1}{2}(u_i - u_{ref,i})^T R (u_i - u_{ref,i})$$

And the following terminal cost:

$$\ell_N(x_N) = \frac{1}{2}(x_N - x_{ref,N})^T Q_f (x_N - x_{ref,N})$$

This is how we will encourange our quadrotor to track a reference trajectory x_{ref}. In the following sections, you will implement `iLQR` and use it inside of a `solve_quadrotor_trajectory` function. Below we have included some starter code, but you are free to use/not use any of the provided functions so long as you pass the tests.

We will consider iLQR to have converged when $\Delta J < \mathrm{atol}$ as calculated during the backwards pass.

In [3]:

```julia
# starter code: feel free to use or not use

function stage_cost(p::NamedTuple,x::Vector,u::Vector,k::Int)
    # TODO: return stage cost at time step k
    Q = p.Q
    R = p.R

    x_ref = p.Xref[k]
    u_ref = p.Uref[k]

    cost = 0.5*(x - x_ref)' * Q * (x - x_ref) + 0.5*(u - u_ref)' * R * (u - u_ref)
    return cost
end
function term_cost(p::NamedTuple,x)
    # TODO: return terminal cost
    Qf = p.Qf

    x_ref = p.Xref[end]

    cost = 0.5*(x - x_ref)' * Qf * (x - x_ref)
    return cost
end
function stage_cost_expansion(p::NamedTuple, x::Vector, u::Vector, k::Int)
    # TODO: return stage cost expansion
    # if the stage cost is J(x,u), you can return the following
    # ∇ₓ²J, ∇ₓJ, ∇ᵤ²J, ∇ᵤJ
    ∇ₓ²J = FD.hessian(x_ -> stage_cost(p, x_, u, k), x)
    ∇ₓJ = FD.gradient(x_ -> stage_cost(p, x_, u, k), x)
    ∇ᵤ²J = FD.hessian(u_ -> stage_cost(p, x, u_, k), u)
    ∇ᵤJ = FD.gradient(u_ -> stage_cost(p, x, u_, k), u)

    return ∇ₓ²J, ∇ₓJ, ∇ᵤ²J, ∇ᵤJ
end
function term_cost_expansion(p::NamedTuple, x::Vector)
    # TODO: return terminal cost expansion
    # if the terminal cost is Jn(x,u), you can return the following
    # ∇ₓ²Jn, ∇ₓJn

    ∇ₓ²J = FD.hessian(x_ -> term_cost(p, x_), x)
    ∇ₓJ = FD.gradient(x_ -> term_cost(p, x_), x)

    return ∇ₓ²J, ∇ₓJ
end
function backward_pass(params::NamedTuple,          # useful params
                       X::Vector{Vector{Float64}},  # state trajectory
                       U::Vector{Vector{Float64}})  # control trajectory
    # compute the iLQR backwards pass given a dynamically feasible trajectory X and U
    # return d, K, ΔJ

    # outputs:
    #     d  - Vector{Vector} feedforward control
    #     K  - Vector{Matrix} feedback gains
    #     ΔJ - Float64        expected decrease in cost

    nx, nu, N = params.nx, params.nu, params.N

    # vectors of vectors/matrices for recursion
    P = [zeros(nx,nx) for i = 1:N]   # cost to go quadratic term
    p = [zeros(nx)    for i = 1:N]   # cost to go linear term
    d = [zeros(nu)    for i = 1:N-1] # feedforward control
    K = [zeros(nu,nx) for i = 1:N-1] # feedback gain

    # TODO: implement backwards pass and return d, K, ΔJ
    N = params.N
    ΔJ = 0.0

    P[N], p[N] = term_cost_expansion(params, X[N])
    Q = params.Q
    R = params.R
    Qf = params.Qf

    for k = N-1:-1:1
        # TODO: compute the feedback and feedforward terms
        A = FD.jacobian(x -> discrete_dynamics(params, x, U[k], k), X[k])
        B = FD.jacobian(u -> discrete_dynamics(params, X[k], u, k), U[k])

        ∇ₓ²J, ∇ₓJ, ∇ᵤ²J, ∇ᵤJ = stage_cost_expansion(params, X[k], U[k], k)
        gx = ∇ₓJ + A'*p[k+1]
        gu = ∇ᵤJ + B'*p[k+1]

        Gxx = ∇ₓ²J + A'*P[k+1]*A
```

```julia
        Guu = ∇ᵤ²J + B'*P[k+1]*B
        Gxu = A'*P[k+1]*B
        Gux = B'*P[k+1]*A

        # Compute feedback and feedforward terms
        d[k] = Guu\gu
        K[k] .= Guu\Gux

        # Update cost-to-go
        p[k] = gx - K[k]'*gu + K[k]'*Guu*d[k] - Gxu*d[k]
        P[k] .= Gxx + K[k]'*Guu*K[k] - Gxu*K[k] - K[k]'*Gux

        # Expected cost reduction
        ΔJ += gu'*d[k]
    end

    return d, K, ΔJ
end

function trajectory_cost(params::NamedTuple,          # useful params
                         X::Vector{Vector{Float64}},  # state trajectory
                         U::Vector{Vector{Float64}}) # control trajectory
    # compute the trajectory cost for trajectory X and U (assuming they are dynamically feasible)
    N = params.N

    cost = 0.0
    # TODO: add trajectory cost
    for k = 1:N-1
        cost += stage_cost(params, X[k], U[k], k)
    end
    cost += term_cost(params, X[N])

    return cost
end

function forward_pass(params::NamedTuple,           # useful params
                      X::Vector{Vector{Float64}},   # state trajectory
                      U::Vector{Vector{Float64}},   # control trajectory
                      d::Vector{Vector{Float64}},   # feedforward controls
                      K::Vector{Matrix{Float64}};   # feedback gains
                      max_linesearch_iters = 20)    # max iters on linesearch
    # forward pass in iLQR with linesearch
    # use a line search where the trajectory cost simply has to decrease (no Armijo)

    # outputs:
    #     Xn::Vector{Vector}  updated state trajectory
    #     Un::Vector{Vector}  updated control trajectory
    #     J::Float64          updated cost
    #     α::Float64.         step length

    nx, nu, N = params.nx, params.nu, params.N

    Xn = [zeros(nx) for i = 1:N]      # new state history
    Un = [zeros(nu) for i = 1:N-1]    # new control history

    # initial condition
    Xn[1] = 1*X[1]

    # initial step length
    α = 1.0

    # TODO: add forward pass
    for i = 1:max_linesearch_iters

        # Try current step size α
        for k = 1:N-1
            # Apply feedback + feedforward control
            δx = Xn[k] - X[k]
            Un[k] = U[k] - α*d[k] - K[k]*δx

            # Roll out dynamics
            Xn[k+1] = discrete_dynamics(params, Xn[k], Un[k], k)
        end

        # Compute new trajectory cost
        Jn = trajectory_cost(params, Xn, Un)
        J = trajectory_cost(params, X, U)

        # If cost decreased, accept the step
        if Jn < J
            return Xn, Un, Jn, α
        end
```

```julia
        # Otherwise reduce step size and reset trajectories
        α *= 0.5
        Xn[1] = X[1] # Reset initial state

    end

    error("forward pass failed")
end
```

```
forward_pass (generic function with 1 method)
```

In [4]:
```julia
function iLQR(params::NamedTuple,         # useful params for costs/dynamics/indexing
              x0::Vector,                 # initial condition
              U::Vector{Vector{Float64}}; # initial controls
              atol=1e-3,                  # convergence criteria: ΔJ < atol
              max_iters = 250,            # max iLQR iterations
              verbose = true)             # print logging

    # iLQR solver given an initial condition x0, initial controls U, and a
    # dynamics function described by `discrete_dynamics`

    # return (X, U, K) where
    # outputs:
    #     X::Vector{Vector} - state trajectory
    #     U::Vector{Vector} - control trajectory
    #     K::Vector{Matrix} - feedback gains K

    # first check the sizes of everything
    @assert length(U) == params.N-1
    @assert length(U[1]) == params.nu
    @assert length(x0) == params.nx

    nx, nu, N = params.nx, params.nu, params.N

    # TODO: initial rollout
    # Initial rollout from x0 using initial controls U
    X = [zeros(nx) for k = 1:N]
    X[1] = x0

    # Forward simulate using initial control sequence
    for k = 1:N-1
        X[k+1] = discrete_dynamics(params, X[k], U[k], k)
    end

    for ilqr_iter = 1:max_iters
        # backward pass
        d, K, ΔJ = backward_pass(params, X, U)
        # forward pass with line search
        X, U, J, α = forward_pass(params, X, U, d, K)

        # termination criteria
        if ΔJ < atol
            if verbose
                @info "iLQR converged"
            end
            return X, U, K
        end

        # ---------------logging -------------------
        if verbose
            dmax = maximum(norm.(d))
            if rem(ilqr_iter-1,10)==0
                @printf "iter     J           ΔJ        |d|         α         \n"
                @printf "-------------------------------------------------\n"
            end
            @printf("%3d   %10.3e  %9.2e  %9.2e  %6.4f    \n",
              ilqr_iter, J, ΔJ, dmax, α)
        end

    end
    error("iLQR failed")
end
```

```
iLQR (generic function with 1 method)
```

In [5]:
```julia
function create_reference(N, dt)
    # create reference trajectory for quadrotor
    R = 6
    Xref = [ [R*cos(t);R*cos(t)*sin(t);1.2 + sin(t);zeros(9)] for t = range(-pi/2,3*pi/2, length = N)]
    for i = 1:(N-1)
        Xref[i][4:6] = (Xref[i+1][1:3] - Xref[i][1:3])/dt
    end
    Xref[N][4:6] = Xref[N-1][4:6]
    Uref = [(9.81*0.5/4)*ones(4) for i = 1:(N-1)]
```

```julia
    return Xref, Uref
end
function solve_quadrotor_trajectory(;verbose = true)

    # problem size
    nx = 12
    nu = 4
    dt = 0.05
    tf = 5
    t_vec = 0:dt:tf
    N = length(t_vec)

    # create reference trajectory
    Xref, Uref = create_reference(N, dt)

    # tracking cost function
    Q = 1*diagm([1*ones(3);.1*ones(3);1*ones(3);.1*ones(3)])
    R = .1*diagm(ones(nu))
    Qf = 10*Q

    # dynamics parameters (these are estimated)
    model = (mass=0.5,
            J=Diagonal([0.0023, 0.0023, 0.004]),
            gravity=[0,0,-9.81],
            L=0.1750,
            kf=1.0,
            km=0.0245,dt = dt)


    # the params needed by iLQR
    params = (
        N = N,
        nx = nx,
        nu = nu,
        Xref = Xref,
        Uref = Uref,
        Q = Q,
        R = R,
        Qf = Qf,
        model = model
    )

    # initial condition
    x0 = 1*Xref[1]

    # initial guess controls
    U = [(uref + .0001*randn(nu)) for uref in Uref]

    # solve with iLQR
    X, U, K = iLQR(params,x0,U;atol=1e-4,max_iters = 250,verbose = verbose)

    return X, U, K, t_vec, params
end
```

```
solve_quadrotor_trajectory (generic function with 1 method)
```

In [6]:

```julia
@testset "ilqr" begin

    # NOTE: set verbose to true here when you submit
    Xilqr, Uilqr, Kilqr, t_vec, params =  solve_quadrotor_trajectory(verbose = true)

    # ---------------testing------------------
    Usol = load(joinpath(@__DIR__,"utils","ilqr_U.jld2"))["Usol"]
    @test maximum(norm.(Usol .- Uilqr,Inf)) <= 1e-2

    # ---------------plotting-----------------
    Xm = hcat(Xilqr...)
    Um = hcat(Uilqr...)
    display(plot(t_vec, Xm[1:3,:]', xlabel = "time (s)", ylabel = "position (m)",
                                title = "Position", label = ["r_x" "r_y" "r_z"]))
    display(plot(t_vec, Xm[4:6,:]', xlabel = "time (s)", ylabel = "velocity (m/s)",
                                title = "Velocity", label = ["v_x" "v_y" "v_z"]))
    display(plot(t_vec, Xm[7:9,:]', xlabel = "time (s)", ylabel = "MRP",
                                title = "Attitude (MRP)", label = ["p_x" "p_y" "p_z"]))
    display(plot(t_vec, Xm[10:12,:]', xlabel = "time (s)", ylabel = "angular velocity (rad/s)",
                                title = "Angular Velocity", label = ["ω_x" "ω_y" "ω_z"]))
    display(plot(t_vec[1:end-1], Um', xlabel = "time (s)", ylabel = "rotor speeds (rad/s)",
                                title = "Controls", label = ["u_1" "u_2" "u_3" "u_4"]))
    display(animate_quadrotor(Xilqr, params.Xref, params.model.dt))
end
```

```
iter     J           ΔJ        |d|         α
------------------------------------------------
  1    3.013e+02   1.34e+05   2.81e+01  1.0000
  2    1.078e+02   5.36e+02   1.34e+01  0.5000
  3    4.906e+01   1.34e+02   4.72e+00  1.0000
  4    4.429e+01   1.16e+01   2.44e+00  1.0000
  5    4.402e+01   8.25e-01   2.54e-01  1.0000
  6    4.398e+01   1.50e-01   8.84e-02  1.0000
  7    4.396e+01   3.99e-02   7.50e-02  1.0000
  8    4.396e+01   1.38e-02   3.90e-02  1.0000
  9    4.396e+01   5.46e-03   3.32e-02  1.0000
 10    4.396e+01   2.48e-03   2.03e-02  1.0000
iter     J           ΔJ        |d|         α
------------------------------------------------
 11    4.396e+01   1.25e-03   1.69e-02  1.0000
 12    4.395e+01   6.83e-04   1.14e-02  1.0000
 13    4.395e+01   4.00e-04   9.37e-03  1.0000
 14    4.395e+01   2.45e-04   6.94e-03  1.0000
 15    4.395e+01   1.55e-04   5.66e-03  1.0000
```












```
┌ Info: iLQR converged
└ @ Main c:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW3_S25\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X21sZmlsZQ==.jl:43
┌ Info: Listening on: 127.0.0.1:8702, thread id: 1
└ @ HTTP.Servers C:\Users\barat\.julia\packages\HTTP\4AUPl\src\Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8702
└ @ MeshCat C:\Users\barat\.julia\packages\MeshCat\9QrxD\src\visualizer.jl:43
```

```
Test Summary: | Pass  Total   Time
ilqr          |    1      1  35.6s
Test.DefaultTestSet("ilqr", Any[], 1, false, false, true, 1.742824473108e9, 1.742824508679e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW3_S25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X23sZmlsZQ==.jl")
```

Part B: Tracking solution with TVLQR (5 pts)

Here we will do the same thing we did in Q1 where we take a trajectory from a trajectory optimization solver, and track it with TVLQR to account for some model mismatch. In DIRCOL, we had to explicitly compute the TVLQR control gains, but in iLQR, we get these same gains out of the algorithmn as the K's. Use these to track the quadrotor through this manuever.

In [7]:

```
@testset "iLQR with model error" begin

    # set verbose to false when you submit
    Xilqr, Uilqr, Kilqr, t_vec, params =  solve_quadrotor_trajectory(verbose = false)

    # real model parameters for dynamics
    model_real = (mass=0.5,
            J=Diagonal([0.0025, 0.002, 0.0045]),
            gravity=[0,0,-9.81],
            L=0.1550,
            kf=0.9,
            km=0.0365,dt = 0.05)

    # simulate closed loop system
    nx, nu, N = params.nx, params.nu, params.N
    Xsim = [zeros(nx) for i = 1:N]
    Usim = [zeros(nx) for i = 1:(N-1)]

    # initial condition
    Xsim[1] = 1*Xilqr[1]

    # TODO: simulate with closed loop control
    for i = 1:(N-1)
        Usim[i] = Uilqr[i] - Kilqr[i]*(Xsim[i] - Xilqr[i])
        Xsim[i+1] = rk4(model_real, quadrotor_dynamics, Xsim[i], Usim[i], model_real.dt)
    end


    # -----------------testing--------------------
    @test 1e-6 <= norm(Xilqr[50] - Xsim[50],Inf) <= .3
    @test 1e-6 <= norm(Xilqr[end] - Xsim[end],Inf) <= .3

    # -----------------plotting--------------------
    Xm = hcat(Xsim...)
    Um = hcat(Usim...)
    Xilqrm = hcat(Xilqr...)
    Uilqrm = hcat(Uilqr...)
    plot(t_vec,Xilqrm[1:3,:]',ls=:dash, label = "",lc = [:red :green :blue])
    display(plot!(t_vec,Xm[1:3,:]',title = "Position (-- is iLQR reference)",
```

```
                 xlabel = "time(s)", ylabel = "position (m)",
                 label = ["r_x" "r_y" "r_z"],lc = [:red :green :blue]))

    plot(t_vec,Xilqrm[7:9,:]',ls=:dash, label = "",lc = [:red :green :blue])
    display(plot!(t_vec,Xm[7:9,:]',title = "Attitude (-- is iLQR reference)",
                 xlabel = "time(s)", ylabel = "Attitude (MRP)",
                 label = ["p_x" "p_y" "p_z"],lc = [:red :green :blue]))

    display(animate_quadrotor(Xilqr, params.Xref, params.model.dt))
end
```






```
┌ Info: Listening on: 127.0.0.1:8703, thread id: 1
└ @ HTTP.Servers C:\Users\barat\.julia\packages\HTTP\4AUPl\src\Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8703
└ @ MeshCat C:\Users\barat\.julia\packages\MeshCat\9QrxD\src\visualizer.jl:43
```

```
Test Summary:          | Pass  Total  Time
iLQR with model error |    2      2  1.2s
Test.DefaultTestSet("iLQR with model error", Any[], 2, false, false, true, 1.74282450908e9, 1.742824510261e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW3_S25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X25sZmlsZQ==.jl")
```

Processing math: 100%

In [1]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
import MeshCat as mc
using Statistics
```

```
  Activating project at `c:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW3_S25`
```

In [2]:
```
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","planar_quadrotor.jl"))
```

```
check_dynamic_feasibility (generic function with 1 method)
```

Q3: Quadrotor Reorientation (40 pts)

In this problem, you will use the trajectory optimization tools you have demonstrated in questions one and two to solve for a collision free reorientation of three planar quadrotors. The planar quadrotor (as described in lecture 10) is described with the following state and dynamics:

$$x = \begin{bmatrix} p_x \\ p_z \\ \theta \\ v_x \\ v_z \\ \omega \end{bmatrix}, \qquad \dot{x} = \begin{bmatrix} v_x \\ v_z \\ \omega \\ \frac{1}{m}(u_1 + u_2)\sin\theta \\ \frac{1}{m}(u_1 + u_2)\cos\theta \\ \frac{\ell}{2J}(u_2 - u_1) \end{bmatrix}$$

where p_x and p_z are the horizontal and vertial positions, v_x and v_z are the corresponding velocities, θ for orientation, ω for angular velocity, ℓ for length of the quadrotor, m for mass, g for gravity acceleration in the $-z$ direction, and a moment of inertia of J.

You are free to use any solver/cost/constraint you would like to solve for three collision free, dynamically feasible trajectories for these quadrotors that looks something like the following:

No description has been provided for this image

(if an animation doesn't load here, check out `quadrotor_reorient.gif`.)

Here are the performance requirements that the resulting trajectories must meet:

- The three quadrotors must start at `x1ic`, `x2ic`, and `x2ic` as shown in the code (these are the initial conditions).
- The three quadrotors must finish their trajectories within **.2** meters of `x1g`, `x2g`, and `x2g` (these are the goal states).
- The three quadrotors must never be within **0.8** meters of one another (use $[p_x, p_z]$ for this).

There are two main ways of going about this:

1. **Cost Shaping**: Design cost functions for each quadrotor that motivates them to take paths that do not result in a collision. You can do something like designing a reference trajectory for each quadrotor to use in the cost. You can use iLQR or DIRCOL for this.
2. **Collision Constraints**: You can optimize over all three quadrotors at once by creating a new state $\tilde{x} = [x_1^T, x_2^T, x_3^T]^T$ and control $\tilde{u} = [u_1^T, u_2^T, u_3^T]^T$, and then directly include collision avoidance constraints. In order to use constraints, you must use DIRCOL (at least for now).

Hints

- You should not use `norm() >= R` in any constraints, instead you should square the constraint to be `norm()^2 >= R^2`. This second constraint is still non-convex, but it is differentiable everywhere.
- If you are using DIRCOL, you can initialize the solver with a "guess" solution by linearly interpolating between the initial and terminal conditions. Julia let's you create a length N linear interpolated vector of vectors between `a::Vector` and `b::Vector` like this: `range(a, b, length = N)` (experiment with this to see how it works).

You can use either RK4 (iLQR or DIRCOL) or Hermite-Simpson (DIRCOL) for your integration. The `dt = 0.2`, and `tf = 5.0` are given for you in the code (you may change these but only if you feel you really have to).

In [3]:
```
function single_quad_dynamics(params, x,u)
    # planar quadrotor dynamics for a single quadrotor

    # unpack state
    px,pz,θ,vx,vz,ω = x

    xdot = [
        vx,
        vz,
        ω,
        (1/params.mass)*(u[1] + u[2])*sin(θ),
        (1/params.mass)*(u[1] + u[2])*cos(θ) - params.g,
        (params.ℓ/(2*params.J))*(u[2]-u[1])
    ]

    return xdot
end
function combined_dynamics(params, x,u)
```

```julia
    # dynamics for three planar quadrotors, assuming the state is stacked
    # in the following manner: x = [x1;x2;x3]

    # NOTE: you would only need to use this if you chose option 2 where
    # you optimize over all three trajectories simultaneously

    # quadrotor 1
    x1 = x[1:6]
    u1 = u[1:2]
    xdot1 = single_quad_dynamics(params, x1, u1)

    # quadrotor 2
    x2 = x[(1:6) .+ 6]
    u2 = u[(1:2) .+ 2]
    xdot2 = single_quad_dynamics(params, x2, u2)

    # quadrotor 3
    x3 = x[(1:6) .+ 12]
    u3 = u[(1:2) .+ 4]
    xdot3 = single_quad_dynamics(params, x3, u3)

    # return stacked dynamics
    return [xdot1;xdot2;xdot3]
end
```

combined_dynamics (generic function with 1 method)

In [4]:

```julia
function create_idx(nx,nu,N)
    # This function creates some useful indexing tools for Z
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]

    # Feel free to use/not use anything here.


    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

"""
    quadrotor_reorient

Function for returning collision free trajectories for 3 quadrotors.

Outputs:
    x1::Vector{Vector}  # state trajectory for quad 1
    x2::Vector{Vector}  # state trajectory for quad 2
    x3::Vector{Vector}  # state trajectory for quad 3
    u1::Vector{Vector}  # control trajectory for quad 1
    u2::Vector{Vector}  # control trajectory for quad 2
    u3::Vector{Vector}  # control trajectory for quad 3
    t_vec::Vector
    params::NamedTuple

The resulting trajectories should have dt=0.2, tf = 5.0, N = 26
where all the x's are length 26, and the u's are length 25.

Each trajectory for quad k should start at `xkic`, and should finish near
`xkg`. The distances between each quad should be greater than 0.8 meters at
every knot point in the trajectory.
"""

function hermite_simpson(params::NamedTuple, x1::Vector, x2::Vector, u, dt::Real)::Vector
    ẋ1 = combined_dynamics(params, x1, u)
    ẋ2 = combined_dynamics(params, x2, u)
    xm = (1/2)*(x1 + x2) + (dt/8)*(ẋ1 - ẋ2)
    ẋm = combined_dynamics(params, xm, u)
    x1 + (dt/6)*(ẋ1 + 4*ẋm + ẋ2) - x2
end

function quadrotor_cost(params::NamedTuple, Z::Vector)::Real
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf

    J = 0
    for i = 1:(N-1)
```

```julia
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]

        J += 0.5*(xi - xg)'*Q*(xi - xg)
        J += 0.5*ui'*R*ui
    end

    xn = Z[idx.x[N]]
    J += 0.5*(xn - xg)'*Qf*(xn - xg)

    return J
end

function dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt

    c = zeros(eltype(Z), idx.nc)

    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]
        xip1 = Z[idx.x[i+1]]
        c[idx.c[i]] = hermite_simpson(params, xi, xip1, ui, dt)
    end
    return c
end

function equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg
    [
        Z[idx.x[1]] - xic;
        Z[idx.x[N]] - xg;
        dynamics_constraints(params, Z)
    ]
end

function inequality_constraint(params, Z)
    c = zeros(eltype(Z), params.N*3)
    idx = params.idx
    off = 0
    for i = 1:params.N
        x1 = Z[idx.x[i][1:2]]
        x2 = Z[idx.x[i][(1:2) .+ 6]]
        x3 = Z[idx.x[i][(1:2) .+ 12]]

        c[(1:3) .+ off] = [
            norm(x1 - x2)^2,
            norm(x1 - x3)^2,
            norm(x2 - x3)^2
        ]
        off += 3
    end
    return c
end

function quadrotor_reorient(;verbose=true)
    nx = 18
    nu = 6
    dt = 0.2
    tf = 5.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    Q = diagm(ones(nx))
    R = 1*diagm(ones(nu))
    Qf = 1*diagm(ones(nx))

    idx = create_idx(nx,nu,N)

    # initial conditions and goal states
    lo = 0.5
    mid = 2
    hi = 3.5
    x1ic = [-2,lo,0,0,0,0]  # ic for quad 1
    x2ic = [-2,mid,0,0,0,0] # ic for quad 2
    x3ic = [-2,hi,0,0,0,0]  # ic for quad 3
    xic = [x1ic;x2ic;x3ic]

    x1g = [2,mid,0,0,0,0]   # goal for quad 1
    x2g = [2,hi,0,0,0,0]    # goal for quad 2
    x3g = [2,lo,0,0,0,0]    # goal for quad 3
    xg = [x1g;x2g;x3g]
    # load all useful things into params
```

```
    # TODO: include anything you would need for a cost function (like a Q, R, Qf if you were doing an
    # LQR cost)
    params = (Q = Q, R = R, Qf = Qf,
              x1ic=x1ic,
              x2ic=x2ic,
              x3ic=x3ic,
              x1g = x1g,
              x2g = x2g,
              x3g = x3g,
              xic = xic, xg = xg,
              dt = dt,
              N = N,
              idx = idx,
              mass = 1.0, # quadrotor mass
              g = 9.81,   # gravity
              ℓ = 0.3,    # quadrotor length
              J = .018,)

    #primal bounds
    x_l = -Inf*ones(idx.nz)
    x_u = Inf*ones(idx.nz)

    #inequality constraint bounds
    c_l = (0.8^2)*ones(3*params.N)
    c_u = Inf*ones(3*params.N)

    #Initial guess
    z0 = zeros(idx.nz)
    x_initialize = range(xic, xg, length = N)
    for i = 1:N
        z0[idx.x[i]] .= x_initialize[i]
    end
    z0 += 0.01*randn(idx.nz)

    # diff type
    diff_type = :auto

    Z = fmincon(quadrotor_cost,equality_constraint,inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = verbose)

    # extrtact solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    # TODO: solve for the three collision free trajectories however you like

    # return the trajectories
    x1 = [x[1:6]          for x in X]
    x2 = [x[(1:6) .+ 6]   for x in X]
    x3 = [x[(1:6) .+ 12]  for x in X]
    u1 = [u[(1:2)]        for u in U]
    u2 = [u[(1:2) .+ 2]   for u in U]
    u3 = [u[(1:2) .+ 4]   for u in U]

    return x1, x2, x3, u1, u2, u3, t_vec, params
end
```

```
quadrotor_reorient (generic function with 1 method)
```

In [5]:
```
@testset "quadrotor reorient" begin

    X1, X2, X3, U1, U2, U3, t_vec, params  = quadrotor_reorient(verbose=true)


    #--------------testing-----------------
    # check lengths of everything
    @test length(X1) == length(X2) == length(X3)
    @test length(U1) == length(U2) == length(U3)
    @test length(X1) == params.N
    @test length(U1) == (params.N-1)

    # check for collisions
    distances = [distance_between_quads(x1[1:2],x2[1:2],x3[1:2]) for (x1,x2,x3) in zip(X1,X2,X3)]
    @test minimum(minimum.(distances)) >= 0.799

    # check initial and final conditions
    @test norm(X1[1] - params.x1ic, Inf) <= 1e-3
    @test norm(X2[1] - params.x2ic, Inf) <= 1e-3
    @test norm(X3[1] - params.x3ic, Inf) <= 1e-3
    @test norm(X1[end] - params.x1g, Inf) <= 2e-1
    @test norm(X2[end] - params.x2g, Inf) <= 2e-1
    @test norm(X3[end] - params.x3g, Inf) <= 2e-1
```

```
    # check dynamic feasibility
    @test check_dynamic_feasibility(params,X1,U1)
    @test check_dynamic_feasibility(params,X2,U2)
    @test check_dynamic_feasibility(params,X3,U3)


    #---------------plotting/animation-------
    display(animate_planar_quadrotors(X1,X2,X3, params.dt))

    plot(t_vec, 0.8*ones(params.N),ls = :dash, color = :red, label = "collision distance",
         xlabel = "time (s)", ylabel = "distance (m)", title = "Distance between Quadrotors")
    display(plot!(t_vec, hcat(distances...)', label = ["|r_1 - r_2|" "|r_1 - r_3|" "|r_2 - r_2|"]))

    X1m = hcat(X1...)
    X2m = hcat(X2...)
    X3m = hcat(X3...)

    plot(X1m[1,:], X1m[2,:], color = :red,title = "Quadrotor Trajectories", label = "quad 1")
    plot!(X2m[1,:], X2m[2,:], color = :green, label = "quad 2",xlabel = "p_x", ylabel = "p_z")
    display(plot!(X3m[1,:], X3m[2,:], color = :blue, label = "quad 3"))

    plot(t_vec, X1m[3,:], color = :red,title = "Quadrotor Orientations", label = "quad 1")
    plot!(t_vec, X2m[3,:], color = :green, label = "quad 2",xlabel = "time (s)", ylabel = "θ")
    display(plot!(t_vec, X3m[3,:], color = :blue, label = "quad 3"))


end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------

******************************************************************************
This program contains Ipopt, a library for large-scale nonlinear optimization.
 Ipopt is released as open source code under the Eclipse Public License (EPL).
         For more information visit https://github.com/coin-or/Ipopt
******************************************************************************

This is Ipopt version 3.14.17, running with linear solver MUMPS 5.7.3.

Number of nonzeros in equality constraint Jacobian...:   300348
Number of nonzeros in inequality constraint Jacobian.:    48204
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      618
                     variables with only lower bounds:        0
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:      486
Total number of inequality constraints...............:       78
        inequality constraints with only lower bounds:       78
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.7157141e+02 1.99e+00 1.43e+00   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  2.7295695e+02 1.99e+00 1.01e+01  -5.7 1.26e+02    -  4.73e-03 2.32e-03h  1
   2  2.7787845e+02 1.98e+00 2.62e+01  -5.8 6.84e+01    -  1.01e-02 6.39e-03h  1
   3  2.8005339e+02 1.97e+00 3.44e+01  -1.4 2.92e+01    -  5.15e-02 4.45e-03h  1
   4  2.8062791e+02 1.96e+00 1.13e+02  -5.8 9.16e+01    -  3.56e-03 1.18e-03h  1
   5  2.8107950e+02 1.96e+00 8.08e+01  -0.4 3.76e+01    -  8.54e-04 1.82e-03f  1
   6  2.8449547e+02 1.95e+00 6.23e+01  -0.9 5.07e+01    -  2.94e-03 6.75e-03h  1
   7  2.9573669e+02 1.91e+00 6.99e+01  -0.2 3.21e+01    -  1.03e-02 1.85e-02h  1
   8  4.3385765e+02 1.79e+00 1.35e+02  -0.7 1.88e+01    -  5.55e-02 1.29e-01h  1
   9  4.6311777e+02 1.69e+00 1.36e+02  -5.8 7.44e+00    -  9.86e-02 5.90e-02h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  8.3242450e+02 1.29e+00 1.74e+02  -2.0 8.10e+00    -  1.44e-01 3.71e-01h  1
  11  8.2418401e+02 1.28e+00 1.72e+02  -5.9 4.16e+01    -  6.07e-02 1.15e-02f  1
  12  8.2218714e+02 1.24e+00 1.69e+02  -0.7 3.20e+01    -  4.94e-02 3.16e-02f  1
  13  8.2652453e+02 1.22e+00 1.66e+02  -1.9 2.05e+01    -  6.70e-02 1.11e-02h  1
  14  9.3267302e+02 1.18e+00 1.36e+02  -0.1 2.07e+01    -  2.43e-01 1.27e-01h  1
  15  2.0297539e+03 1.04e+00 3.93e+01  -1.1 3.59e+00    -  6.06e-01 1.00e+00h  1
  16  2.0193604e+03 2.87e-01 1.58e+01  -1.7 1.27e+00    -  5.79e-01 1.00e+00f  1
  17  2.0343734e+03 2.35e-01 1.28e+01  -1.9 4.25e-01    -  6.19e-01 1.79e-01h  1
  18  2.0980060e+03 2.62e-02 8.65e+00  -1.7 9.40e-01    -  2.41e-01 9.57e-01h  1
  19  2.0986644e+03 2.09e-02 7.03e+00  -1.5 5.34e-01    -  3.94e-01 2.05e-01h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  2.0994731e+03 1.55e-02 6.16e+00  -2.7 4.13e-01    -  3.98e-01 2.62e-01h  1
  21  2.1019798e+03 1.37e-03 4.94e+00  -3.1 4.81e-01    -  1.50e-01 1.00e+00h  1
  22  2.1017367e+03 9.27e-04 3.25e+00  -3.0 5.31e-01    -  3.69e-01 3.38e-01h  1
```

```
  23  2.1014750e+03 8.72e-04 3.83e+00  -2.8 2.58e+00    -  2.13e-01 7.83e-02f  1
  24  2.1012470e+03 6.24e-04 2.96e+00  -2.5 1.43e+00    -  1.32e-01 2.35e-01f  1
  25  2.1012131e+03 2.92e-04 2.84e+00  -2.8 2.02e-01    -  1.00e+00 5.22e-01h  1
  26  2.1012145e+03 9.13e-06 1.64e-01  -3.9 2.25e-02    -  1.00e+00 1.00e+00h  1
  27  2.1012143e+03 1.62e-06 9.70e-02  -5.6 7.04e-03    -  1.00e+00 9.85e-01h  1
  28  2.1012149e+03 1.20e-07 3.72e-02  -7.5 2.77e-03    -  1.00e+00 9.94e-01h  1
  29  2.1012149e+03 1.21e-08 1.57e-02  -9.1 1.29e-03    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  2.1012149e+03 4.86e-09 1.84e-03 -11.0 4.58e-04    -  1.00e+00 1.00e+00h  1
  31  2.1012149e+03 3.52e-10 9.02e-05 -11.0 1.40e-04    -  1.00e+00 1.00e+00h  1
  32  2.1012149e+03 4.65e-11 8.26e-06 -11.0 3.25e-05    -  1.00e+00 1.00e+00h  1
  33  2.1012149e+03 1.45e-12 2.47e-06 -11.0 1.06e-05    -  1.00e+00 1.00e+00h  1
  34  2.1012149e+03 6.33e-14 8.41e-07 -11.0 4.81e-06    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 34

                                   (scaled)                 (unscaled)
Objective...............:   2.1012148825700415e+03    2.1012148825700415e+03
Dual infeasibility......:   8.4095573482656505e-07    8.4095573482656505e-07
Constraint violation....:   6.3282712403633923e-14    6.3282712403633923e-14
Variable bound violation:   0.0000000000000000e+00    0.0000000000000000e+00
Complementarity.........:   1.0000173637594265e-11    1.0000173637594265e-11
Overall NLP error.......:   8.4095573482656505e-07    8.4095573482656505e-07


Number of objective function evaluations             = 35
Number of objective gradient evaluations             = 35
Number of equality constraint evaluations            = 35
Number of inequality constraint evaluations          = 35
Number of equality constraint Jacobian evaluations   = 35
Number of inequality constraint Jacobian evaluations = 35
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 8.446

EXIT: Optimal Solution Found.
┌ Info: Listening on: 127.0.0.1:8700, thread id: 1
└ @ HTTP.Servers C:\Users\barat\.julia\packages\HTTP\4AUPl\src\Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8700
└ @ MeshCat C:\Users\barat\.julia\packages\MeshCat\9QrxD\src\visualizer.jl:43
```


Distance between Quadrotors
distance (m)
time (s)
collision distance
|r_1 - r_2|
|r_1 - r_3|
|r_2 - r_2|


Quadrotor Trajectories
p_z
p_x
quad 1
quad 2
quad 3




```
Test Summary:      | Pass  Total   Time
quadrotor reorient |   14     14  25.0s
Test.DefaultTestSet("quadrotor reorient", Any[], 14, false, false, true, 1.742824473417e9, 1.742824498374e9, fal
se, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW3_S25\\jl_notebo
ok_cell_df34fa98e69747e1a8f8a730347b8e2f_W5sZmlsZQ==.jl")
```

Processing math: 100%