In [1]:
```julia
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import FiniteDiff
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
using MeshCat
const mc = MeshCat
using StaticArrays
using Printf
```

```
  Activating project at `c:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW4_S25`
┌ Warning: The active manifest file has dependencies that were resolved with a different julia version (1.10.7). Unexpected behavior may occur.
└ @ nothing C:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW4_S25\Manifest.toml:0
```

In [2]:
```julia
include(joinpath(@__DIR__, "utils","ilc_visualizer.jl"))
```

```
update_car_pose! (generic function with 1 method)
```

Q1: Iterative Learning Control (ILC) (40 pts)

In this problem, you will use ILC to generate a control trajectory for a Car as it swerves to avoid a moose, also known as "the moose test" (wikipedia, video). We will model the dynamics of the car as with a simple nonlinear bicycle model, with the following state and control:

$$x = \begin{bmatrix} p_x \\ p_y \\ \theta \\ \delta \\ v \end{bmatrix}, \qquad u = \begin{bmatrix} a \\ \dot{\delta} \end{bmatrix}$$

where p_x and p_y describe the 2d position of the bike, θ is the orientation, δ is the steering angle, and v is the velocity. The controls for the bike are acceleration a, and steering angle rate $\dot{\delta}$.

In [3]:
```julia
function estimated_car_dynamics(model::NamedTuple, x::Vector, u::Vector)::Vector
    # nonlinear bicycle model continuous time dynamics
    px, py, θ, δ, v = x
    a, δdot = u

    β = atan(model.lr * δ, model.L)
    s,c = sincos(θ + β)
    ω = v*cos(β)*tan(δ) / model.L

    vx = v*c
    vy = v*s

    xdot = [
        vx,
        vy,
        ω,
        δdot,
        a
    ]

    return xdot
end
function rk4(model::NamedTuple, ode::Function, x::Vector, u::Vector, dt::Real)::Vector
    k1 = dt * ode(model, x,        u)
    k2 = dt * ode(model, x + k1/2, u)
    k3 = dt * ode(model, x + k2/2, u)
    k4 = dt * ode(model, x + k3,   u)
    return x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end
```

```
rk4 (generic function with 1 method)
```

We have computed an optimal trajectory X_{ref} and U_{ref} for a moose test trajectory offline using this `estimated_car_dynamics` function. Unfortunately, this is a highly approximate dynamics model, and when we run U_{ref} on the car, we get a very different trajectory than we expect. This is caused by a significant sim to real gap. Here we will show what happens when we run these controls on the true dynamics:

In [4]:
```julia
function load_car_trajectory()
    # load in trajectory we computed offline
    path = joinpath(@__DIR__, "utils","init_control_car_ilc.jld2")
    F = jldopen(path)
    Xref = F["X"]
    Uref = F["U"]
    close(F)
    return Xref, Uref
end
function true_car_dynamics(model::NamedTuple, x::Vector, u::Vector)::Vector
    # true car dynamics
    px, py, θ, δ, v = x
    a, δdot = u

    # sluggish controls (not in the approximate version)
    a = 0.9*a - 0.1
    δdot = 0.9*δdot - .1*δ + .1

    β = atan(model.lr * δ, model.L)
    s,c = sincos(θ + β)
    ω = v*cos(β)*tan(δ) / model.L

    vx = v*c
    vy = v*s

    xdot = [
        vx,
        vy,
        ω,
        δdot,
        a
    ]

    return xdot
end

@testset "sim to real gap" begin
    # problem size
    nx = 5
    nu = 2
    dt = 0.1
    tf = 5.0
    t_vec = 0:dt:tf
    N = length(t_vec)
    model = (L = 2.8, lr = 1.6)

    # optimal trajectory computed offline with approximate model
    Xref, Uref = load_car_trajectory()

    # TODO: simulated Uref with the true car dynamics and store the states in Xsim
    Xsim = [zeros(nx) for i = 1:N]
    Xsim[1] = Xref[1]
    for i=1:N-1
        u = Uref[i]
        Xsim[i+1] = rk4(model, true_car_dynamics, Xsim[i], u, dt)
    end

    # --------testing------------------
    @test norm(Xsim[1] - Xref[1]) == 0
    @test norm(Xsim[end] - [3.26801052, 15.0590156, 2.0482790, 0.39056168, 4.5],Inf) < 1e-4

    # --------plotting/animation-------
    Xm= hcat(Xsim...)
    Xrefm = hcat(Xref...)
    plot(Xrefm[1,:], Xrefm[2,:], ls = :dash, label = "reference",
         xlabel = "x (m)", ylabel = "y (m)", title = "Simulation vs Reference")
    display(plot!(Xm[1,:], Xm[2,:], label = "actual"))

end
```




```
Test Summary:   | Pass  Total   Time
sim to real gap |    2      2  12.7s
Test.DefaultTestSet("sim to real gap", Any[], 2, false, false, true, 1.744500187383e9, 1.744500200112e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW4_S25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_W5sZmlsZQ==.jl")
```

In order to account for this, we are going to use ILC to iteratively correct our control until we converge.

To encourage the trajectory of the bike to follow the reference, the objective value for this problem is the following:

$$J(X, U) = \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{ref,i})^T Q (x_i - x_{ref,i}) + \frac{1}{2}(u_i - u_{ref,i})^T R (u_i - u_{ref,i}) \right] + \frac{1}{2}(x_N - x_{ref,N})^T Q_f (x_N - x_{ref,N})$$

Using ILC as described in Lecture 18, we are to linearize our approximate dynamics model about X_{ref} and U_{ref} to get the following Jacobians:

$$A_k = \frac{\partial f}{\partial x}\bigg|_{x_{ref,k}, u_{ref,k}}, \qquad B_k = \frac{\partial f}{\partial u}\bigg|_{x_{ref,k}, u_{ref,k}}$$

where $f(x, u)$ is our **approximate discrete** dynamics model (`estimated_car_dynamics` + `rk4`). **You will form these Jacobians exactly once, using `Xref` and `Uref` .** Here is a summary of the notation:

- X_{ref} (`Xref`) - Optimal trajectory computed offline with approximate dynamics model.
- U_{ref} (`Uref`) - Optimal controls computed offline with approximate dynamics model.
- X_{sim} (`Xsim`) - Simulated trajectory with real dynamics model.
- $\bar{U}$ (`Ubar`) - Control we use for simulation with real dynamics model (this is what ILC updates).

In the second step of ILC, we solve the following optimization problem:

$$\begin{aligned} \min_{\Delta x_{1:N}, \Delta u_{1:N-1}} \quad & J(X_{sim} + \Delta X, \bar{U} + \Delta U) \\ \text{st} \quad & \Delta x_1 = 0 \\ & \Delta x_{k+1} = A_k \Delta x_k + B_k \Delta u_k \quad \text{for } k = 1, 2, \ldots, N-1 \end{aligned}$$

We are going to initialize our $\bar{U}$ with U_{ref}, then the ILC algorithm will update $\bar{U} = \bar{U} + \Delta U$ at each iteration. It should only take 5-10 iterations to converge down to $\|$. You do not need to do any sort of linesearch between ILC updates.

In [5]:
```julia
# feel free to use/not use any of these

function trajectory_cost(Xsim::Vector{Vector{Float64}}, # simulated states
                         Ubar::Vector{Vector{Float64}}, # simulated controls (ILC iterates this)
                         Xref::Vector{Vector{Float64}}, # reference X's we want to track
                         Uref::Vector{Vector{Float64}}, # reference U's we want to track
                         Q::Matrix,                     # LQR tracking cost term
                         R::Matrix,                     # LQR tracking cost term
                         Qf::Matrix                     # LQR tracking cost term
                         )::Float64                     # return cost J

    J = 0
    # TODO: return trajectory cost J(Xsim, Ubar)
    for i=1:length(Xsim)-1
        J += (Xsim[i] - Xref[i])' * Q * (Xsim[i] - Xref[i])
        J += (Ubar[i] - Uref[i])' * R * (Ubar[i] - Uref[i])
    end
    J += (Xsim[end] - Xref[end])' * Qf * (Xsim[end] - Xref[end])
```

```julia
    return J
end

function vec_from_mat(Xm::Matrix)::Vector{Vector{Float64}}
    # convert a matrix into a vector of vectors
    X = [Xm[:,i] for i = 1:size(Xm,2)]
    return X
end

function ilc_update(Xsim::Vector{Vector{Float64}}, # simulated states
                    Ubar::Vector{Vector{Float64}}, # simulated controls (ILC iterates this)
                    Xref::Vector{Vector{Float64}}, # reference X's we want to track
                    Uref::Vector{Vector{Float64}}, # reference U's we want to track
                    As::Vector{Matrix{Float64}},   # vector of A jacobians at each time step
                    Bs::Vector{Matrix{Float64}},   # vector of B jacobians at each time step
                    Q::Matrix,                     # LQR tracking cost term
                    R::Matrix,                     # LQR tracking cost term
                    Qf::Matrix                     # LQR tracking cost term
                    )::Vector{Vector{Float64}}     # return vector of ΔU's

    # solve optimization problem for ILC update
    N = length(Xsim)
    nx,nu = size(Bs[1])

    # create variables
    ΔX = cvx.Variable(nx, N)
    ΔU = cvx.Variable(nu, N-1)

    # TODO: cost function (tracking cost on Xref, Uref)
    cost = 0.0
    for i = 1:N-1
        ex = Xsim[i] - Xref[i]
        eu = Ubar[i] - Uref[i]
        cost += 0.5*cvx.quadform(ΔX[:, i], Q) + ex' * Q * ΔX[:, i] +
                0.5*cvx.quadform(ΔU[:, i], R) + eu' * R * ΔU[:, i]
    end
    exN = Xsim[end] - Xref[end]
    cost += 0.5*cvx.quadform(ΔX[:, end], Qf) + exN' * Qf * ΔX[:, end]

    # TODO: initial condition constraint
    cons = [ΔX[:,1] == zeros(nx)]

    # TODO: dynamics constraints
    for i = 1:N-1
        push!(cons, ΔX[:, i+1] == As[i]*ΔX[:, i] + Bs[i]*ΔU[:, i])
    end

    # problem instance
    prob = cvx.minimize(cost, cons)

    cvx.solve!(prob, ECOS.Optimizer; silent = true)

    # return ΔU
    ΔU = vec_from_mat(ΔU.value)

    return ΔU
end
```

```
ilc_update (generic function with 1 method)
```

Here you will run your ILC algorithm. The resulting plots should show the simulated trajectory `Xsim` tracks `Xref` very closely, but there should be a significant difference between `Uref` and `Ubar`.

In [6]:

```julia
@testset "ILC" begin

    # problem size
    nx = 5
    nu = 2
    dt = 0.1
    tf = 5.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # optimal trajectory computed offline with approximate model
    Xref, Uref = load_car_trajectory()

    # initial and terminal conditions
    xic = Xref[1]
    xg = Xref[N]

    # LQR tracking cost to be used in ILC
    Q = diagm([1,1,.1,.1,.1])
```

```julia
R = .1*diagm(ones(nu))
Qf = 1*diagm(ones(nx))

# load all useful things into params
model = (L = 2.8, lr = 1.6)

params = (Q = Q, R = R, Qf = Qf,xic = xic, xg = xg, Xref=Xref,Uref=Uref,
      dt = dt,
      N = N,
      model = model)


# this holds the sim trajectory (with real dynamics)
Xsim = [zeros(nx) for i = 1:N]

# this is the feedforward control ILC is updating
Ubar = [zeros(nu) for i = 1:(N-1)]
Ubar .= Uref # initialize Ubar with Uref

# TODO: calculate Jacobians
As = [zeros(nx, nx) for i = 1:(N-1)]
Bs = [zeros(nx, nu) for i = 1:(N-1)]
for i = 1:(N-1)
    As[i] = FD.jacobian(x -> rk4(model, estimated_car_dynamics, x, Uref[i], dt), Xref[i])
    Bs[i] = FD.jacobian(u -> rk4(model, estimated_car_dynamics, Xref[i], u, dt), Uref[i])
end

# logging stuff
@printf "iter     objv        |ΔU|     \n"
@printf "------------------------------\n"

for ilc_iter = 1:10 # it should not take more than 10 iterations to converge

    # TODO: rollout
    Xsim = [zeros(nx) for i = 1:N]
    Xsim[1] = xic
    for i = 1:(N-1)
        u = Ubar[i]
        Xsim[i+1] = rk4(model, true_car_dynamics, Xsim[i], u, dt)
    end

    # TODO: calculate objective val (trajectory_cost)
    obj_val = trajectory_cost(Xsim, Ubar, Xref, Uref, Q, R, Qf)

    # solve optimization problem for update (ilc_update)
    ΔU = ilc_update(Xsim, Ubar, Xref, Uref, As, Bs, Q, R, Qf)

    # TODO: update the control
    for i = 1:(N-1)
        Ubar[i] += ΔU[i]
    end

    # logging
    @printf("%3d   %10.3e  %10.3e  \n", ilc_iter, obj_val, sum(norm.(ΔU)))

end

# --------plotting/animation-------
Xm= hcat(Xsim...)
Um = hcat(Ubar...)
Xrefm = hcat(Xref...)
Urefm = hcat(Uref...)
plot(Xrefm[1,:], Xrefm[2,:], ls = :dash, label = "reference",
     xlabel = "x (m)", ylabel = "y (m)", title = "Trajectory")
display(plot!(Xm[1,:], Xm[2,:], label = "actual"))

plot(t_vec[1:end-1], Urefm', ls = :dash, lc = [:green :blue],label = "",
     xlabel = "time (s)", ylabel = "controls", title = "Controls (-- is reference)")
display(plot!(t_vec[1:end-1], Um', label = ["δ̇" "a"], lc = [:green :blue]))

# animation
vis = Visualizer()
vis_traj!(vis, :traj, [[x[1],x[2],0.1] for x in Xsim]; R = 0.02)
build_car!(vis[:car])
anim = mc.Animation(vis, fps=floor(Int,1/dt))
for k = 1:N
    mc.atframe(anim, k) do
        update_car_pose!(vis[:car], Xsim[k])
    end
end
mc.setanimation!(vis, anim)
display(render(vis))
```

```
    # ----------testing------------
    @test 0.1 <= sum(norm.(Xsim - Xref)) <= 1.0 # should be ~0.7
    @test 5 <=  sum(norm.(Ubar - Uref)) <= 10 # should be ~7.7

end
```

```
iter     objv        |ΔU|
------------------------------
  1    2.872e+03   6.307e+01
  2    2.262e+03   4.498e+01
  3    1.024e+03   9.266e+01
  4    1.222e+01   1.394e+01
  5    9.228e-01   1.959e+00
  6    1.549e-01   1.679e-01
  7    1.431e-01   1.649e-02
  8    1.429e-01   1.574e-03
  9    1.429e-01   1.981e-04
 10    1.429e-01   2.738e-05
```






```
┌ Info: Listening on: 127.0.0.1:8700, thread id: 1
└ @ HTTP.Servers C:\Users\barat\.julia\packages\HTTP\4AUPl\src\Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8700
└ @ MeshCat C:\Users\barat\.julia\packages\MeshCat\9QrxD\src\visualizer.jl:43
```

```
Test Summary: | Pass  Total      Time
ILC           |    2      2  1m05.5s
Test.DefaultTestSet("ILC", Any[], 2, false, false, true, 1.744500201433e9, 1.744500266948e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW4_S25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X12sZmlsZQ==.jl")
```

Loading [MathJax]/jax/element/mml/optable/GeneralPunctuation.js

In [1]:
```julia
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
using MeshCat
const mc = MeshCat
using StaticArrays
using Printf
```

```
  Activating project at `c:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW4_S25`
┌ Warning: The active manifest file has dependencies that were resolved with a different julia version (1.10.7). Unexpected behavior may occur.
└ @ nothing C:\CMU\SEM II\OCRL\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\HW4_S25\Manifest.toml:0
```

Julia note:

incorrect:

`x_l[idx.x[i]][2] = 0 # this does not change x_l`

correct:

`x_l[idx.x[i][2]] = 0 # this changes x_l`

It should always be `v[index] = new_val` if I want to update v with `new_val` at `index`.

In [2]:
```julia
let

    # vector we want to modify
    Z = randn(5)

    # original value of Z so we can check if we are changing it
    Z_original = 1 * Z

    # index range we are considering
    idx_x = 1:3

    # this does NOT change Z
    Z[idx_x][2] = 0

    # we can prove this
    @show norm(Z - Z_original)

    # this DOES change Z
    Z[idx_x[2]] = 0

    # we can prove this
    @show norm(Z - Z_original)


end
```

```
norm(Z - Z_original) = 0.0
norm(Z - Z_original) = 0.33364723680539177
0.33364723680539177
```

In [3]:
```julia
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","walker.jl"))
```

```
update_walker_pose! (generic function with 1 method)
```

(If nothing loads here, check out `walker.gif` in the repo)

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q2: Hybrid Trajectory Optimization (60 pts)

In this problem you'll use a direct method to optimize a walking trajectory for a simple biped model, using the hybrid dynamics formulation. You'll pre-specify a gait sequence and solve the problem using Ipopt. Your final solution should look like the video above.

The Dynamics

Our system is modeled as three point masses: one for the body and one for each foot. The state is defined as the x and y positions and velocities of these masses, for a total of 6 degrees of freedom and 12 states. We will label the position and velocity of each body with the following notation:

$$r^{(b)} = \begin{bmatrix} p_x^{(b)} \\ p_y^{(b)} \end{bmatrix} \qquad v^{(b)} = \begin{bmatrix} v_x^{(b)} \\ v_y^{(b)} \end{bmatrix}$$

$$r^{(1)} = \begin{bmatrix} p_x^{(1)} \\ p_y^{(1)} \end{bmatrix} \qquad v^{(1)} = \begin{bmatrix} v_x^{(1)} \\ v_y^{(1)} \end{bmatrix}$$

$$r^{(2)} = \begin{bmatrix} p_x^{(2)} \\ p_y^{(2)} \end{bmatrix} \qquad v^{(2)} = \begin{bmatrix} v_x^{(2)} \\ v_y^{(2)} \end{bmatrix}$$

Each leg is connected to the body with prismatic joints. The system has three control inputs: a force along each leg, and the torque between the legs.

The state and control vectors are ordered as follows:

$$x = \begin{bmatrix} p_x^{(b)} \\ p_y^{(b)} \\ p_x^{(1)} \\ p_y^{(1)} \\ p_x^{(2)} \\ p_y^{(2)} \\ v_x^{(b)} \\ v_y^{(b)} \\ v_x^{(1)} \\ v_y^{(1)} \\ v_x^{(2)} \\ v_y^{(2)} \end{bmatrix} \qquad u = \begin{bmatrix} F^{(1)} \\ F^{(2)} \\ \tau \end{bmatrix}$$

where e.g. $p_x^{(b)}$ is the x position of the body, $v_y^{(i)}$ is the y velocity of foot i, $F^{(i)}$ is the force along leg i, and τ is the torque between the legs.

The continuous time dynamics and jump maps for the two stances are shown below:

In [4]:
```
function stance1_dynamics(model::NamedTuple, x::Vector, u::Vector)
    # dynamics when foot 1 is in contact with the ground

    mb,mf = model.mb, model.mf
    g = model.g

    M = Diagonal([mb mb mf mf mf mf])

    rb  = x[1:2]   # position of the body
    rf1 = x[3:4]   # position of foot 1
    rf2 = x[5:6]   # position of foot 2
    v   = x[7:12]  # velocities
```

```
    ℓ1x = (rb[1]-rf1[1])/norm(rb-rf1)
    ℓ1y = (rb[2]-rf1[2])/norm(rb-rf1)
    ℓ2x = (rb[1]-rf2[1])/norm(rb-rf2)
    ℓ2y = (rb[2]-rf2[2])/norm(rb-rf2)

    B = [ℓ1x  ℓ2x  ℓ1y-ℓ2y;
         ℓ1y  ℓ2y  ℓ2x-ℓ1x;
          0    0     0;
          0    0     0;
          0  -ℓ2x  ℓ2y;
          0  -ℓ2y -ℓ2x]

    v̇ = [0; -g; 0; 0; 0; -g] + M\(B*u)

    ẋ = [v; v̇]

    return ẋ
end

function stance2_dynamics(model::NamedTuple, x::Vector, u::Vector)
    # dynamics when foot 2 is in contact with the ground

    mb,mf = model.mb, model.mf
    g = model.g
    M = Diagonal([mb mb mf mf mf mf])

    rb  = x[1:2]   # position of the body
    rf1 = x[3:4]   # position of foot 1
    rf2 = x[5:6]   # position of foot 2
    v   = x[7:12]  # velocities

    ℓ1x = (rb[1]-rf1[1])/norm(rb-rf1)
    ℓ1y = (rb[2]-rf1[2])/norm(rb-rf1)
    ℓ2x = (rb[1]-rf2[1])/norm(rb-rf2)
    ℓ2y = (rb[2]-rf2[2])/norm(rb-rf2)

    B = [ℓ1x  ℓ2x  ℓ1y-ℓ2y;
         ℓ1y  ℓ2y  ℓ2x-ℓ1x;
        -ℓ1x   0   -ℓ1y;
        -ℓ1y   0    ℓ1x;
          0    0     0;
          0    0     0]

    v̇ = [0; -g; 0; -g; 0; 0] + M\(B*u)

    ẋ = [v; v̇]

    return ẋ
end

function jump1_map(x)
    # foot 1 experiences inelastic collision
    xn = [x[1:8]; 0.0; 0.0; x[11:12]]
    return xn
end

function jump2_map(x)
    # foot 2 experiences inelastic collision
    xn = [x[1:10]; 0.0; 0.0]
    return xn
end

function rk4(model::NamedTuple, ode::Function, x::Vector, u::Vector, dt::Real)::Vector
    k1 = dt * ode(model, x,        u)
    k2 = dt * ode(model, x + k1/2, u)
    k3 = dt * ode(model, x + k2/2, u)
    k4 = dt * ode(model, x + k3,   u)
    return x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end
```

```
rk4 (generic function with 1 method)
```

We are setting up this problem by scheduling out the contact sequence. To do this, we will define the following sets:

$$\mathcal{M}_1 = \{1:5, 11:15, 21:25, 31:35, 41:45\}$$
$$\mathcal{M}_2 = \{6:10, 16:20, 26:30, 36:40\}$$

where $\mathcal{M}_1$ contains the time steps when foot 1 is pinned to the ground (`stance1_dynamics`), and $\mathcal{M}_2$ contains the time steps when foot 2 is pinned to the ground (`stance2_dynamics`). The jump map sets $\mathcal{J}_1$ and $\mathcal{J}_2$ are the indices where the mode of the next time step is different than the current, i.e. $\mathcal{J}_i \equiv \{k+1 \notin \mathcal{M}_i \mid k \in \mathcal{M}_i\}$. We can write these out explicitly as the following:

$$\mathcal{J}_1 = \{5, 15, 25, 35\}$$
$$\mathcal{J}_2 = \{10, 20, 30, 40\}$$

Another term you will see is set subtraction, or $\mathcal{M}_i \setminus \mathcal{J}_i$. This just means that if $k \in \mathcal{M}_i \setminus \mathcal{J}_i$, then k is in $\mathcal{M}_i$ but not in $\mathcal{J}_i$.

We will make use of the following Julia code for determining which set an index belongs to:

In [5]:
```
let
    M1 = vcat([ (i-1)*10      .+ (1:5)   for i = 1:5]...) # stack the set into a vector
    M2 = vcat([((i-1)*10 + 5) .+ (1:5)   for i = 1:4]...) # stack the set into a vector
    J1 = [5,15,25,35]
    J2 = [10,20,30,40]

    @show  (5 in M1) # show if 5 is in M1
    @show  (5 in J1) # show if 5 is in J1
    @show !(5 in M1) # show is 5 is not in M1

    @show (5 in M1) && !(5 in J1) # 5 in M1 but not J1 (5 ∈ M_1 \ J1)

end
```

```
5 in M1 = true
5 in J1 = true
!(5 in M1) = false
5 in M1 && !(5 in J1) = false
false
```

We are now going to setup and solve a constrained nonlinear program. The optimization problem looks complicated but each piece should make sense and be relatively straightforward to implement. First we have the following LQR cost function that will track x_{ref} (`Xref`) and u_{ref} (`Uref`):

$$J(x_{1:N}, u_{1:N-1}) = \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{ref,i})^T Q(x_i - x_{ref,i}) + \frac{1}{2}(u_i - u_{ref,i})^T R(u_i - u_{ref,i})\right] + \frac{1}{2}(x_N - x_{ref,N})^T Q_f(x_N - x_{ref,N})$$

Which goes into the following full optimization problem: $\min$

Each constraint is now described, with the type of constraint for `fmincon` in parantheses:

1. Initial condition constraint **(equality constraint)**.
2. Terminal condition constraint **(equality constraint)**.
3. Stance 1 discrete dynamics **(equality constraint)**.
4. Stance 2 discrete dynamics **(equality constraint)**.
5. Discrete dynamics from stance 1 to stance 2 with jump 2 map **(equality constraint)**.
6. Discrete dynamics from stance 2 to stance 1 with jump 1 map **(equality constraint)**.
7. Make sure the foot 1 is pinned to the ground in stance 1 **(equality constraint)**.
8. Make sure the foot 2 is pinned to the ground in stance 2 **(equality constraint)**.
9. Length constraints between main body and foot 1 **(inequality constraint)**.
10. Length constraints between main body and foot 2 **(inequality constraint)**.
11. Keep the y position of all 3 bodies above ground **(primal bound)**.

And here we have the list of mathematical functions to the Julia function names:

- f_1 is `stance1_dynamics` + `rk4`
- f_2 is `stance2_dynamics` + `rk4`
- g_1 is `jump1_map`
- g_2 is `jump2_map`

For instance, g_2(f_1(x_k,u_k)) is `jump2_map(rk4(model, stance1_dynamics, xk, uk, dt))`

Remember that r^{(b)} is defined above.

In [6]:
```
function reference_trajectory(model, xic, xg, dt, N)
    # creates a reference Xref and Uref for walker

    Uref = [[model.mb*model.g*0.5;model.mb*model.g*0.5;0] for i = 1:(N-1)]

    Xref = [zeros(12) for i = 1:N]

    horiz_v = (3/N)/dt
    xs = range(-1.5, 1.5, length = N)
    Xref[1] = 1*xic
    Xref[N] = 1*xg

    for i = 2:(N-1)
        Xref[i] = [xs[i],1,xs[i],0,xs[i],0,horiz_v,0,horiz_v,0,horiz_v,0]
    end
```

```julia
    return Xref, Uref
end
```

```
reference_trajectory (generic function with 1 method)
```

To solve this problem with Ipopt and `fmincon`, we are going to concatenate all of our x's and u's into one vector (same as HW3Q1):

$$Z = \begin{bmatrix}x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N \cdot nx + (N-1)\cdot nu}$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z. Remember that the API for `fmincon` (that we used in HW3Q1) is the following: $\begin{align} \min_{z} \quad & \ell(z) & \text{cost function}\\ \text{st} \quad & c_{eq}(z) = 0 & \text{equality constraint}\\ & c_L \leq c_{ineq}(z) \leq c_U & \text{inequality constraint}\\ & z_L \leq z \leq z_U & \text{primal bound constraint} \end{align}$

Template code has been given to solve this problem but you should feel free to do whatever is easiest for you, as long as you get the trajectory shown in the animation `walker.gif` and pass tests.

In [7]:
```julia
# feel free to solve this problem however you like, below is a template for a
# good way to start.

function create_idx(nx,nu,N)
    # create idx for indexing convenience
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]
    # and stacked dynamics constraints of size nx are
    # c[idx.c[i]] = <dynamics constraint at time step i>
    #
    # feel free to use/not use this

    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function walker_cost(params::NamedTuple, Z::Vector)::Real
    # cost function
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf
    Xref,Uref = params.Xref, params.Uref

    # TODO: input walker LQR cost

    J = 0
    for i = 1:(N-1)
        xi   = Z[idx.x[i]]       # current state
        ui   = Z[idx.u[i]]       # current control
        x_ref = Xref[i]          # reference state at time i
        u_ref = Uref[i]          # reference control at time i
        J += 0.5 * (xi - x_ref)' * Q * (xi - x_ref)
        J += 0.5 * (ui - u_ref)' * R * (ui - u_ref)
    end

    xN    = Z[idx.x[N]]
    x_ref_N = Xref[N]
    J += 0.5 * (xN - x_ref_N)' * Qf * (xN - x_ref_N)
    return J


    return J
end

function walker_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2
    J1, J2 = params.J1, params.J2
    model = params.model

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)
    nx = params.idx.nx

    # TODO: input walker dynamics constraints (constraints 3-6 in the opti problem)
    for k = 1:(N-1)
        xk = Z[idx.x[k]]
```

```julia
        uk = Z[idx.u[k]]
        xkp1 = Z[idx.x[k+1]]
        # Determine which dynamics to use based on k
        # (Assume each time step belongs either to M1 or M2)
        if k in M1
            # In stance 1
            if k in J1
                # Jump from stance1 to stance2: apply jump map 2 on f1.
                x_next_est = jump2_map(rk4(model, stance1_dynamics, xk, uk, dt))
            else
                # Normal stance 1 dynamics.
                x_next_est = rk4(model, stance1_dynamics, xk, uk, dt)
            end
        elseif k in M2
            # In stance 2
            if k in J2
                # Jump from stance2 to stance1: apply jump map 1 on f2.
                x_next_est = jump1_map(rk4(model, stance2_dynamics, xk, uk, dt))
            else
                # Normal stance 2 dynamics.
                x_next_est = rk4(model, stance2_dynamics, xk, uk, dt)
            end
        else
            error("Time index k=$(k) not in either M1 or M2")
        end
        # Impose the dynamics constraint: x_{k+1} - x_next_est = 0.
        # Place the nx residuals into the proper block of c.
        c[(k-1)*nx+1 : k*nx] = xkp1 - x_next_est
    end
    return c
end

function walker_stance_constraint(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2
    J1, J2 = params.J1, params.J2

    model = params.model

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), N)

    # TODO: add walker stance constraints (constraints 7-8 in the opti problem)
    for k = 1:N
        xk = Z[idx.x[k]]
        if k in M1
            c[k] = xk[4]    # foot 1's y (should be 0)
        elseif k in M2
            c[k] = xk[6]    # foot 2's y (should be 0)
        else
            # Ideally every time step is in one of the sets.
            c[k] = 0.0
        end
    end
    return c
end

function walker_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # TODO: stack up all of our equality constraints

    # should be length 2*nx + (N-1)*nx + N
    # inital condition constraint (nx)       (constraint 1)
    # terminal constraint         (nx)       (constraint 2)
    # dynamics constraints        (N-1)*nx   (constraint 3-6)
    # stance constraint           N          (constraint 7-8)

    # Initial condition:
    ceq_ic = Z[idx.x[1]] - xic
    # Terminal condition:
    ceq_term = Z[idx.x[N]] - xg
    # Dynamics constraints:
    ceq_dyn = walker_dynamics_constraints(params, Z)
    # Stance constraints:
    ceq_stance = walker_stance_constraint(params, Z)

    return vcat(ceq_ic, ceq_term, ceq_dyn, ceq_stance)
end

function walker_inequality_constraint(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2
```

```julia
    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), 2*N)

    # TODO: add the length constraints shown in constraints (9-10)
    # there are 2*N constraints here
    for k = 1:N
        xk = Z[idx.x[k]]
        # Extract positions from x_k:
        r_b = xk[1:2]
        r_1 = xk[3:4]
        r_2 = xk[5:6]
        d1 = norm(r_b - r_1)
        d2 = norm(r_b - r_2)
        c[2*k - 1] = d1
        c[2*k    ] = d2
    end
    return c
end
```

walker_inequality_constraint (generic function with 1 method)

In [8]:

```julia
@testset "walker trajectory optimization" begin

    # dynamics parameters
    model = (g = 9.81, mb= 5.0, mf = 1.0, ℓ_min = 0.5, ℓ_max = 1.5)

    # problem size
    nx = 12
    nu = 3
    tf = 4.4
    dt = 0.1
    t_vec = 0:dt:tf
    N = length(t_vec)

    # initial and goal states
    xic = [-1.5;1;-1.5;0;-1.5;0;0;0;0;0;0;0]
    xg =  [1.5;1;1.5;0;1.5;0;0;0;0;0;0;0]

    # index sets
    M1 = vcat([ (i-1)*10      .+ (1:5)   for i = 1:5]...)
    M2 = vcat([((i-1)*10 + 5) .+ (1:5)   for i = 1:4]...)
    J1 = [5,15,25,35]
    J2 = [10,20,30,40]

    # reference trajectory
    Xref, Uref = reference_trajectory(model, xic, xg, dt, N)

    # LQR cost function (tracking Xref, Uref)
    Q = diagm([1; 10; fill(1.0, 4); 1; 10; fill(1.0, 4)]);
    R = diagm(fill(1e-3,3))
    Qf = 1*Q;

    # create indexing utilities
    idx = create_idx(nx,nu,N)

    # put everything useful in params
    params = (
        model = model,
        nx = nx,
        nu = nu,
        tf = tf,
        dt = dt,
        t_vec = t_vec,
        N = N,
        M1 = M1,
        M2 = M2,
        J1 = J1,
        J2 = J2,
        xic = xic,
        xg = xg,
        idx = idx,
        Q = Q, R = R, Qf = Qf,
        Xref = Xref,
        Uref = Uref
    )

    # TODO: primal bounds (constraint 11)
    x_l = -Inf*ones(idx.nz)
    x_u =  Inf*ones(idx.nz)
    for i = 1:N
        state_inds = idx.x[i]      # indices for x_i in Z
        x_l[state_inds[2]] = 0.0
        x_l[state_inds[4]] = 0.0
```

```julia
        x_l[state_inds[6]] = 0.0
    end


    # TODO: inequality constraint bounds
    c_l = -Inf*ones(2*N) # update this
    c_u =  Inf*ones(2*N) # update this

    # TODO: initialize z0 with the reference Xref, Uref
    z0 = zeros(idx.nz)
    for i = 1:(N-1)
        z0[idx.x[i]] = Xref[i]
        z0[idx.u[i]] = Uref[i]
    end

    # adding a little noise to the initial guess is a good idea
    z0 = z0 + (1e-6)*randn(idx.nz)

    diff_type = :auto

    Z = fmincon(walker_cost,walker_equality_constraint,walker_inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    # ------------plotting-------------
    Xm = hcat(X...)
    Um = hcat(U...)

    plot(Xm[1,:],Xm[2,:], label = "body")
    plot!(Xm[3,:],Xm[4,:], label = "leg 1")
    display(plot!(Xm[5,:],Xm[6,:], label = "leg 2",xlabel = "x (m)",
                  ylabel = "y (m)", title = "Body Positions"))

    display(plot(t_vec[1:end-1], Um',xlabel = "time (s)", ylabel = "U",
                 label = ["F1" "F2" "τ"], title = "Controls"))

    # ---------animation-------------
    vis = Visualizer()
    build_walker!(vis, model::NamedTuple)
    anim = mc.Animation(vis, fps=floor(Int,1/dt))
    for k = 1:N
        mc.atframe(anim, k) do
            update_walker_pose!(vis, model::NamedTuple, X[k])
        end
    end
    mc.setanimation!(vis, anim)
    display(render(vis))

    # --------testing----------------

    # initial and terminal states
    @test norm(X[1] - xic,Inf) <= 1e-3
    @test norm(X[end] - xg,Inf) <= 1e-3

    for x in X

        # distance between bodies
        rb = x[1:2]
        rf1 = x[3:4]
        rf2 = x[5:6]
        @test (0.5 - 1e-3) <= norm(rb-rf1) <= (1.5 + 1e-3)
        @test (0.5 - 1e-3) <= norm(rb-rf2) <= (1.5 + 1e-3)

        # no two feet moving at once
        v1 = x[9:10]
        v2 = x[11:12]
        @test min(norm(v1,Inf),norm(v2,Inf)) <= 1e-3

        # check everything above the surface
        @test x[2] >= (0 - 1e-3)
        @test x[4] >= (0 - 1e-3)
        @test x[6] >= (0 - 1e-3)

    end

end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
```

```
---------testing objective gradient----------------
---------testing constraint Jacobian---------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve---------------------

******************************************************************************
This program contains Ipopt, a library for large-scale nonlinear optimization.
 Ipopt is released as open source code under the Eclipse Public License (EPL).
         For more information visit https://github.com/coin-or/Ipopt
******************************************************************************

This is Ipopt version 3.14.17, running with linear solver MUMPS 5.7.3.

Number of nonzeros in equality constraint Jacobian...:   401184
Number of nonzeros in inequality constraint Jacobian.:    60480
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      672
                     variables with only lower bounds:      135
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:      597
Total number of inequality constraints...............:       90
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  8.2799992e+00 1.50e+00 1.09e+01   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  6.8202724e+02 2.52e+00 2.40e+04  -0.5 1.18e+02    -  1.47e-02 8.94e-01H  1
   2  4.9074710e+02 2.04e+00 2.55e+04  -0.8 5.10e+01    -  1.00e+00 6.81e-01f  1
   3  4.8264088e+02 1.65e+00 2.46e+04   0.4 4.35e+01    -  1.00e+00 2.00e-01h  1
   4  5.2731966e+02 1.64e+00 7.20e+04   1.1 3.17e+01    -  1.00e+00 1.00e+00f  1
   5  5.3556791e+02 7.75e-01 2.57e+05   0.7 1.58e+01    -  1.00e+00 5.30e-01h  1
   6  5.3838143e+02 1.14e+00 1.50e+04  -0.2 9.01e+00    -  1.00e+00 9.19e-01h  1
   7  5.1322923e+02 9.56e-01 5.46e+03  -1.0 9.77e+00    -  1.00e+00 8.59e-01f  1
   8  5.0567987e+02 1.07e-01 3.14e+03  -0.7 1.17e+01    -  1.00e+00 1.00e+00f  1
   9  4.7735642e+02 1.18e+00 1.13e+03  -1.6 5.35e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  4.2585669e+02 8.82e-01 5.58e+02  -1.8 3.25e+01    -  1.00e+00 1.00e+00f  1
  11  3.8400846e+02 6.15e-01 2.03e+02  -1.6 1.76e+01    -  1.00e+00 1.00e+00f  1
  12  3.5458247e+02 2.03e-01 2.52e+02  -1.6 1.30e+01    -  1.00e+00 1.00e+00f  1
  13  3.3865316e+02 1.79e-02 2.38e+02  -2.3 1.57e+01    -  1.00e+00 1.00e+00f  1
  14  3.3850581e+02 2.05e-01 1.45e+03  -1.1 1.83e+02    -  1.00e+00 6.97e-02f  4
  15  3.1661802e+02 1.56e-01 1.77e+02  -1.7 2.45e+01    -  1.00e+00 1.00e+00f  1
  16  3.4258066e+02 4.71e-01 3.34e+03  -0.4 1.04e+02    -  1.00e+00 4.05e-01f  2
  17  3.1503537e+02 4.30e-01 2.59e+01  -0.5 3.56e+01    -  1.00e+00 1.00e+00f  1
  18  2.7079187e+02 1.92e-01 4.73e+01  -0.7 2.39e+01    -  9.94e-01 1.00e+00f  1
  19  2.6911323e+02 5.09e-02 2.68e+01  -1.2 1.32e+01    -  9.99e-01 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  2.5871802e+02 2.34e-02 1.25e+01  -1.6 9.16e+00    -  1.00e+00 1.00e+00f  1
  21  2.5731532e+02 4.08e-03 3.21e+00  -2.5 7.66e+00    -  1.00e+00 1.00e+00f  1
  22  2.5560299e+02 2.39e-03 1.83e+00  -3.5 3.85e+00    -  1.00e+00 1.00e+00f  1
  23  2.5412826e+02 2.48e-03 2.81e+00  -5.0 4.29e+00    -  1.00e+00 1.00e+00f  1
  24  2.5269899e+02 1.26e-03 2.38e+00  -6.5 6.70e+00    -  1.00e+00 1.00e+00f  1
  25  2.5151529e+02 6.93e-03 6.73e+01  -6.7 6.39e+01    -  1.00e+00 1.92e-01f  1
  26  2.5119489e+02 6.65e-03 7.92e+01  -4.1 6.32e+01    -  1.00e+00 2.96e-02f  1
  27  2.5051523e+02 1.09e-02 3.15e+01  -4.3 5.03e+01    -  7.04e-01 2.92e-01f  2
  28  2.4951400e+02 1.69e-02 1.49e+01  -3.7 4.27e+01    -  1.00e+00 3.51e-01f  1
  29  2.5114245e+02 4.72e-03 8.27e+00  -3.1 6.96e+00    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  2.4846851e+02 5.29e-03 1.36e+01  -1.4 7.08e+00    -  9.83e-01 1.00e+00f  1
  31  2.4845229e+02 1.52e-03 1.41e+00  -2.7 2.85e+00    -  1.00e+00 1.00e+00h  1
  32  2.4810284e+02 7.60e-04 6.90e-01  -3.7 1.87e+00    -  1.00e+00 1.00e+00f  1
  33  2.4809165e+02 1.55e-04 5.69e-01  -3.9 1.16e+00    -  1.00e+00 1.00e+00f  1
  34  2.4797920e+02 1.36e-04 3.63e-01  -5.5 1.13e+00    -  1.00e+00 1.00e+00f  1
  35  2.4794159e+02 1.25e-04 2.98e+01  -7.0 2.14e+00    -  1.00e+00 3.87e-01f  1
  36  2.4794118e+02 1.25e-04 1.10e+02  -8.1 6.47e+00    -  1.00e+00 1.73e-03h  1
  37  2.4794067e+02 1.62e-03 1.69e+02  -8.1 3.05e+01    -  1.00e+00 1.93e-01f  1
  38  2.4792563e+02 1.58e-03 2.04e+02  -5.3 1.13e+01    -  1.00e+00 2.93e-02h  1
  39  2.4863926e+02 1.64e-03 6.63e+01 -11.0 1.48e+01    -  4.05e-01 8.91e-01H  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  40  2.4790863e+02 2.22e-03 7.98e-01  -5.8 6.90e+00    -  1.00e+00 1.00e+00f  1
  41  2.4812923e+02 1.65e-03 2.62e+01  -3.7 4.12e+00    -  3.45e-01 1.00e+00h  1
  42  2.4779382e+02 1.05e-03 6.55e+00  -4.0 2.84e+00    -  1.00e+00 6.23e-01f  1
  43  2.4774656e+02 1.17e-04 2.09e-01  -5.7 1.15e+00    -  1.00e+00 1.00e+00h  1
  44  2.4773187e+02 6.54e-05 2.59e+01  -7.3 3.60e-01    -  1.00e+00 5.00e-01h  2
  45  2.4772926e+02 1.46e-05 2.47e+01  -6.4 1.81e-01    -  1.00e+00 8.15e-01h  1
  46  2.4772886e+02 3.63e-06 1.96e+01  -6.5 3.24e-02    -  1.00e+00 7.56e-01h  1
  47  2.4772844e+02 1.16e-06 2.28e-02  -7.9 1.13e-01    -  1.00e+00 1.00e+00h  1
  48  2.4772865e+02 1.00e-08 6.23e-02  -9.0 3.00e-01    -  1.00e+00 1.00e+00H  1
  49  2.4772835e+02 7.99e-07 2.70e+02 -10.8 2.82e-01    -  1.00e+00 2.50e-01f  3
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
```

```
  50  2.4772771e+02 1.07e-06 1.28e-02 -11.0 8.17e-02    -  1.00e+00 1.00e+00h  1
  51  2.4772765e+02 1.38e-07 6.99e-03 -11.0 6.81e-02    -  1.00e+00 1.00e+00h  1
  52  2.4772942e+02 1.44e-08 7.97e-02 -11.0 5.27e-01    -  1.00e+00 1.00e+00H  1
  53  2.4772816e+02 1.96e-06 6.48e-02 -11.0 1.00e-01    -  1.00e+00 1.00e+00f  1
  54  2.4772759e+02 2.80e-06 2.26e-02 -11.0 2.59e-01    -  1.00e+00 1.00e+00h  1
  55  2.4772775e+02 3.57e-07 3.42e-02 -11.0 5.69e-02    -  1.00e+00 1.00e+00h  1
  56  2.4772755e+02 2.06e-07 8.29e-03 -11.0 3.58e-02    -  1.00e+00 1.00e+00h  1
  57  2.4772756e+02 4.56e-08 7.74e-03 -11.0 9.06e-03    -  1.00e+00 1.00e+00h  1
  58  2.4772755e+02 1.87e-08 6.74e-04 -11.0 7.76e-03    -  1.00e+00 1.00e+00h  1
  59  2.4772755e+02 1.00e-08 4.20e-04 -11.0 1.64e-03    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  60  2.4772755e+02 1.99e-08 2.75e-03 -11.0 1.79e-02    -  1.00e+00 1.00e+00h  1
  61  2.4772758e+02 1.00e-08 7.12e-03 -11.0 4.03e-02    -  1.00e+00 1.00e+00H  1
  62  2.4772755e+02 9.49e-08 3.18e-03 -11.0 2.19e-02    -  1.00e+00 1.00e+00h  1
  63  2.4772756e+02 3.88e-08 6.03e-03 -11.0 1.67e-02    -  1.00e+00 1.00e+00h  1
  64  2.4772754e+02 1.38e-08 1.46e-03 -11.0 1.13e-02    -  1.00e+00 1.00e+00h  1
  65  2.4772754e+02 1.00e-08 1.45e-03 -11.0 3.35e-03    -  1.00e+00 1.00e+00h  1
  66  2.4772754e+02 1.00e-08 2.42e-04 -11.0 1.27e-03    -  1.00e+00 1.00e+00h  1
  67  2.4772754e+02 1.00e-08 1.71e-04 -11.0 7.58e-04    -  1.00e+00 1.00e+00h  1
  68  2.4772754e+02 1.00e-08 6.49e-05 -11.0 8.38e-04    -  1.00e+00 1.00e+00h  1
  69  2.4772754e+02 1.00e-08 1.17e-03 -11.0 1.41e-02    -  1.00e+00 1.00e+00H  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  70  2.4772754e+02 1.00e-08 5.62e+02 -11.0 9.77e-03    -  1.00e+00 5.00e-01h  2
  71  2.4772754e+02 1.00e-08 9.75e-04 -11.0 8.65e-03    -  1.00e+00 1.00e+00h  1
  72  2.4772754e+02 1.00e-08 1.13e-03 -11.0 2.36e-03    -  1.00e+00 1.00e+00h  1
  73  2.4772754e+02 1.00e-08 1.80e-04 -11.0 4.06e-03    -  1.00e+00 1.00e+00h  1
  74  2.4772754e+02 1.00e-08 2.63e-04 -11.0 7.66e-04    -  1.00e+00 1.00e+00h  1
  75  2.4772754e+02 1.00e-08 8.99e-06 -11.0 7.03e-04    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 75

                                   (scaled)                 (unscaled)
Objective...............:   2.4772754375903389e+02    2.4772754375903389e+02
Dual infeasibility......:   8.9930430153262186e-06    8.9930430153262186e-06
Constraint violation....:   9.9999997785351066e-09    9.9999997785351066e-09
Variable bound violation:   9.9999997785351066e-09    9.9999997785351066e-09
Complementarity.........:   1.0000000023838081e-11    1.0000000023838081e-11
Overall NLP error.......:   2.0052036071639427e-07    8.9930430153262186e-06


Number of objective function evaluations             = 106
Number of objective gradient evaluations             = 76
Number of equality constraint evaluations            = 106
Number of inequality constraint evaluations          = 106
Number of equality constraint Jacobian evaluations   = 76
Number of inequality constraint Jacobian evaluations = 76
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 47.821

EXIT: Optimal Solution Found.
```






```
┌ Info: Listening on: 127.0.0.1:8701, thread id: 1
└ @ HTTP.Servers C:\Users\barat\.julia\packages\HTTP\4AUPl\src\Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8701
└ @ MeshCat C:\Users\barat\.julia\packages\MeshCat\9QrxD\src\visualizer.jl:43
```

```
Test Summary:                  | Pass  Total     Time
walker trajectory optimization |  272    272  1m17.4s
Test.DefaultTestSet("walker trajectory optimization", Any[], 272, false, false, true, 1.744500191643e9, 1.744500
269049e9, false, "c:\\CMU\\SEM II\\OCRL\\16745---Optimal-Control-and-Reinforcement-Learning---Spring-2025\\HW4_S
25\\jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X16sZmlsZQ==.jl")
```